

武汉纺织大学

《单片机原理及应用》习题集 (第五版李朝青编)

《单片机原理及应用》课程教学团队

2024-11

目 录

第一章习题

第二章习题

第三章习题

第四章习题

第五章习题

第六章习题

第七章习题

期中测验试卷

期末模拟试卷

第一章习题

1. 二进制数 1101B 化成十进制数

解: $1101B = 1 \times 2^3 + 1 \times 2^2 + 0 \times 2^1 + 1 \times 2^0 = 13D$

2. 十六进制数 1FH 化成十进制数

解: $1FH = 1 \times 16^1 + 15 \times 16^0 = 31D$

3. 十进制数 11D 转换成二进制数

解: $11D = 1011B$

$$\begin{array}{r|l} 2 & 11 \text{ 余数} \\ 2 & 5 \quad 1 \\ 2 & 2 \quad 1 \\ 2 & 1 \quad 0 \\ & 0 \quad 1 \end{array}$$

提示: 将十进制数除 2 取余, 商为 0 止余数倒置。

4. 十进制数 100D 转换成十六进制数

解: $100D = 64H$

$$\begin{array}{r|l} 16 & 100 \text{ 余数} \\ 16 & 6 \quad 4 \\ & 0 \quad 6 \end{array}$$

提示: 将十进制数除 16 取余, 商为 0 止余数倒置。

5. 二进制数 100111100B 转换成十六进制数

解: $1\ 0011\ 1100\ B = 0001\ 0011\ 1100\ B = 13C\ H$

提示: 将二进制数以小数点为界四位一分, 不足补 0, 用一位十六进制数代替四位二进制数。

6. 十六进制转换成二进制数

解: $D4E\ H = 1101\ 0100\ 1110\ B$

提示: 将十六进制数以小数点为界, 用四位二进制数代替一位十六进制数。

7. 将十进制 0.625 转换成二进制数。

解: 将十进制数乘 2 取整, 小数部分为 0 止或满足精度, 整数部分正的排列。

$$\begin{array}{ll} 0.625 \times 2 = 1.25 & 1 \text{ (高位)} \\ 0.25 \times 2 = 0.5 & 0 \\ 0.5 \times 2 = 1.00 & 1 \text{ (低位)} \\ 0.625D = 0.101B \end{array}$$

8. 将十进制 0.359375 转换成十六进制数。

解: 将十进制数乘 16 取整, 小数部分为 0 止或满足精度, 整数部分正的排列。

$$\begin{array}{ll} 0.359375 \times 16 = 5.75 & 5 \text{ (高位)} \\ 0.75 \times 16 = 12.0 & 12 \text{ (低位)} \\ 0.359375D = 0.5CH \end{array}$$

9. 将二进制 0.01011011, 0.1011001101 转换成十六进制数。

解:

$$0.01011011\text{B} = 0.0101\ 1011 = 0.5\text{BH}$$

$$0.1011001101\text{B} = 0.1011\ 0011\ 0100 = 0.\text{B}34\text{H}$$

10. 将十六进制 8F.41, 175.4E 转换成二进制数。

解:

$$8\text{F}.41\text{H} = 1000\ 1111 . 0100\ 0001\ \text{B}$$

$$175.4\text{EH} = 0001\ 0111\ 0101 . 0100\ 1110\ \text{B}$$

11. 怎样根据真值求补码, 或根据补码求真值?

答: 只有两种求补码的方法:

1) 是求负数的补码, 用绝对值“取反加 1”来求补码;

2) 是求负数(补码)的真值, 可先将该补码数用“取反加 1”的方法得到其绝对值, 再在绝对值前添加一负号。

12. 将下列各二进制数转换为十进制数及十六进制数。

① 11010B ② 10101011B

解: 11010B=26D

$$10101011\text{B} = 171\text{D}$$

$$11010\text{B} = 1\text{AH}$$

$$10101011\text{B} = \text{ABH}$$

13. 将下列各十进制数转换为十六进制数。

① 129D ② 253D

解: 129D=81H

$$253\text{D} = \text{FDH}$$

14. 将下列各十六进制数转换成二进制数和十进制数。

① 5AH ② 0AE7.D2H

解: 5AH=1011010B

$$0\text{AE}7.\text{D}2\text{H} = 101011100111.1101001\text{B}$$

$$5\text{AH} = 90\text{D}$$

$$0\text{AE}7.\text{D}2\text{H} = 2791.8203125\text{D}$$

15. 将十六进制数 2AFH 用二进制数表示为(B)。

A. 001010100111B B. 001010101111B

C. 001010101011B D. 001010101101B

16. 将二进制数 1111010101B 用十六进制数表示为(C)。

A. 5D3H B. 5C3H C. 3D5H D. 3C5H

第二章习题

1. 求 89C51 单片机外接晶振频率 12 MHz 时的各种时序单位的大小。

解：振荡周期= $1/f_{osc}=1/12\text{ MHz}=0.0833\text{ }\mu\text{s}$

状态周期= $2/f_{osc}=2/12\text{ MHz}=0.167\text{ }\mu\text{s}$

机器周期= $12/f_{osc}=12/12\text{ MHz}=1\text{ }\mu\text{s}$

指令周期=（1~4）机器周期= $1\sim 4\text{ }\mu\text{s}$

2. 89C51 单片机的机器周期为 $2\mu\text{s}$, 则其晶振频率 f_{osc} 为(C)MHz。

A.1 B.2 C.6 D.12

第三章习题

1. 请分析下面指令，说明按顺序执行每条指令后的结果。

```
MOV SP, #18H      ; (SP) =18H
MOV A, #30H        ; (A) =30H
MOV DPTR, #1000H   ; (DPTR) =1000H
PUSH A              ; (SP) =19H (19H) =30H
PUSH DPH            ; (SP) =1AH (1AH) =10H
PUSH DPL            ; (SP) =1BH (1BH) =00H
POP DPL             ; (DPL)=00H   (SP)=1AH
POP DPH             ; (DPH)=10H   (SP)=19H
POP A               ; (A) =30H   (SP)=18H
```

2. 请分析下面指令，说明按顺序执行每条指令后的结果。

```
CLR C              ;CY=0
CLR ACC.0          ;ACC.0=0
CPL ACC.0          ;ACC.0=1
SETB RS1           ;RS1=1
CLR RS0            ;RS0=0 选择当前工作寄存器为第 2 组
CPL C              ;CY=1
```

3. 若：初始 ROM 内容：

2000H	00H
2001H	01H
2002H	09H
2003H	04H
2004H	06H
2005H	08H
2006H	FFH
2007H	66H
2008H	45H
2009H	ABH
200AH	11H

执行下面程序段后结果如何？

```
MOV DPTR, #2000H   ; (DPTR) =2000H
MOV A, #0AH         ; (A) =0AH
MOVC A, @A+DPTR     ; (A) =(200AH)=11H
```

4. 若初始 ROM 内容为：

1010H	01H
1011H	02H
1012H	03H
1013H	04H

执行下面程序后的结果如何？

$$\begin{array}{r} \text{(A)} = 0C9H = 11001001 \text{ B} \\ - \text{) } \quad \text{CY} = 1 = 00000001 \text{ B} \\ \hline \quad \quad \quad 11001000 \text{ B} \\ - \text{) } \quad \text{(R2)} = 54H = 01010100 \text{ B} \\ \hline \quad \quad \quad \text{(A)} = 74H = 01110100 \text{ B} \end{array}$$

结果为: (A) = 74H

标志位为: CY=0 AC=0 OV=1 P=0

10. (A) = 4EH, (B) = 5DH, 执行指令“MUL AB”后结果如何?

解: 结果为: (B) = 1CH, (A) = 56H。

表示积 (BA) = 1C56H, OV=1。

11. 若: (R1) = 30H, (30H) = 11H

求执行下面指令后的结果。

INC @R1; (30H) = 12H

INC R1; (R1) = 31H

解: 结果为: (30H) = 12H, (R1) = 31H。不影响 PSW

13. 1) 1000H: SJMP 55H;

其转移目的地址=?

解: ∵ rel = 55H = 0101 0101B, 为正数。

∴ 目的 PC = 1000H + 2 + rel
= 1002H + 55H
= 1057H

2) 1000H: SJMP F6H;

其转移目的地址=?

解: ∵ rel = F6H = 1111 0110B 为负数, 表示向低地址方向转移。

∴ 目的 PC = 1000H + 2 + rel = 1002H + F6H 不能直接相加, 要先计算出 F6H 的真值。即:

F6H = 1111 0110B 取反加 1 后为: 0000 1010B = 0AH

∴ 目的 PC = 1002H + F6H
= 1002H - 0AH
= 0FF8H

13. 在 89C51 片内 RAM 中, 已知 (30H) = 38H, (38H) = 40H, (40H) = 48H, (48H) = 90H。请分析下面各是什么指令, 说明源操作数的寻址方式以及按顺序执行每条指令后的结果。

MOV A, 40H

MOV R0, A

MOV P1, #0F0H

MOV @R0, 30H

MOV DPTR, #3848H

MOV 40H, 38H

MOV R0, 30H

MOV P0, R0

MOV 18H, #30H

MOV A, @R0

MOV P2, P1

答:

MOV A, 40H; 直接寻址 (40H) → A


```

MOV R0, A ; 寄存器寻址 (A) → R0
MOV P1, #0F0H ; 立即数寻址 0F0H → P1
MOV @R0, 30H ; 直接寻址 (30H) → (R0)
MOV DPTR, #3848H ; 立即数寻址 3848H → DPTR
MOV 40H, 38H ; 直接寻址 (38H) → 40H
MOV R0, 30H ; 直接寻址 (30H) → R0
MOV P0, R0 ; 寄存器寻址 (R0) → P0
MOV 18H, #30H ; 立即数寻址 30H → 18H
MOV A, @R0 ; 寄存器间接寻址 ((R0)) → A
MOV P2, P1 ; 直接寻址 (P1) → P2
最后结果: (R0) = 38H, (A) = 40H, (P0) = 38H, (P1) = (P2) = 0F0H, (DPTR)
= 3848H, (18H) = 30H, (30H) = 38H, (38H) = 40H, (40H) = 40H, (48H) = 38H

```

14. 设 R0 的内容为 32H, A 的内容为 48H, 片内 RAM 的 32H 单元内容为 80H, 40H 单元内容为 08H。请指出在执行下列程序段后上述各单元内容的变化。

```

MOV A, @R0
MOV @R0, 40H
MOV 40H, A
MOV R0, #35H

```

答:

```

MOV A, @R0 ; ((R0)) = 80H → A
MOV @R0, 40H ; (40H) = 08H → (R0)
MOV 40H, A ; (A) = 80 → 40H
MOV R0, #35H ; 35H → R0

```

最后结果: (R0) = 35H (A) = 80H, (32H) = 08H, (40H) = 80H

15. 已知 (A) = 83H, (R0) = 17H, (17H) = 34H。请写出执行完下列程序段后 A 的内容。

```

ANL A, #17H
ORL 17H, A
XRL A, @R0
CPL A

```

答:

```

ANL A, #17H ; 83H ∧ 17H = 03H → A
ORL 17H, A ; 34H ∨ 03H = 37H → 17H
XRL A, @R0 ; 03H ⊕ 37H = 34H
CPL A ; 34H 求反等于 CBH

```

所以 (A) = CBH

16.1) (SP) = 60H, (A) = 59H, (R0) = 68H, 执行指令:

```

PUSH ACC
PUSH R0
POP ACC
POP R0

```

后, (SP) = 60H, (A) = 68H, (R0) = 59H。

2) (SP)=60H, (A)=59H, (R0)=68H, 执行指令:

SS1:

MOV R7, #0

DJNZ R7, SS1

后, (R7)=0, SS1 被执行 无数 次。

3) (SP)=5FH, (50H)=4FH, (40H)=04H, 执行指令:

PUSH 50H

PUSH 40H

POP 40H

POP 50H

后, (50H)=4FH, (40H)=04H。

17. 执行指令: JBC 40H, rel 后, (40H)=0。

18. 89C51 单片机系统频率 f_{osc} 为 6MHz, 子程序 DEL 的功能为 (A)。

解:

DEL: MOV R7, #200

DEL1: MOV R6, #125

DEL2: DJNZ R6, DEL2

DJNZ R7, DEL1

RET

A. 延时约 100ms B. 延时约 50ms C. DEL2 被执行 125 次 D. 以上都不对

19. 1) 设 (A)=85H, (20H)=0AFH, 执行指令 ADD A, 20H, 结果 (B)。

A. (A)=34H, CY=0, AC=1, OV=0 B. (A)=34H, CY=1, AC=1, OV=1

C. (A)=4FH, CY=1, AC=1, OV=0 D. (A)=34H, CY=1, AC=0, OV=0

2) 设 (A)=53H, (R0)=0FCH, 执行指令 ADD A, R0; 结果 (B)。

A. (A)=4FH, CY=0, AC=0, OV=0,

B. (A)=4FH, CY=1, AC=0, OV=0,

C. (A)=34H, CY=1, AC=0, OV=0,

D. (A)=4FH, CY=1, AC=0, OV=1,

20. 1) 执行指令

MOV DPTR, #2003H

MOV A, #18H

MOV 20H, #38H

MOV R0, #20H

XCH A, @R0

后, (A)=38H, (20H)=18H。

2) 执行指令

```

MOV R7, #10H
MOV DPTR, #DATA
MOV B, #0
LOOP: CLR A
      MOVC A, @A+DPTR
      CJNE A,B,NEXT1
NEXT1: JC NEXT2
      MOV B,A
NEXT2: INC DPTR
      DJNZ R7,LOOP
HLT:  SJMP HLT
DATA: DB 01H,43H,06H,0EH,0AH,0DH,0FH
      DB 0EH,0FH,0DH,34H,7EH,3CH,55H,0EH
程序执行后,(B)=7EH, (R7)=0。

```

3) 执行指令

```

MOV  A, #38H
CLR  C
SUBB A, #30H
CJNE A, #0AH, ASC1
ASC1: JC  ASC2
SUBB A, #07H
ASC2: SJMP ASC2
后, A.=08H。

```

21. 执行下列 3 条指令后, 30H 单元的内容是 (**C**)。

```

MOV  R0, #30H
MOV  40H, #0EH
MOV  @R0, 40H
A.40H   B.30H   C.0EH   D.FFH

```

22. A7H 和 5BH 两个立即数相加后, 和为 (), (CY) 是 (**C**)。

A. 1CH, 0 B. C1H, 0
C. 02H, 1 D. C1H, 1

23. 若 (A) = 86H, (PSW) = 80H, 则执行 RRC A 指令后, (A) = (**A**)

A. C3H B. B3H C. 0DH D. 56H

24. 如果 (P0) = 65H, 则当

```

CPL  P0.2
SETB C
MOV  P0.4, C
执行后 (P0) = ( C )
A. 61H    B. 75H    C. 71H    D. 17H

```

25. 假定累加器 A 中的内容为 30H, 执行指令

1000H: MOVC A, @A+PC

后, 把程序存储器 (C) 单元的内容送入累加器 A 中。

A. 1000H B. 1030H C. 1031H D. 1032H

26. 编写程序, 将外部数据存储器中的 4000H-40FFH 单元全部清零。 (A)

解:

```
ORG 0000H
MAIN: MOV A, #0 ; 送预置数给 A
      MOV R0, ( ) ; 设置循环次数
      MOV DPTR, ( ) ; 设置数据指针的初值
LOOP: MOVX @DPTR, A ; 当前单元清零
      INC DPTR ; 指向下一个单元
      DJNZ R0, LOOP ; 是否结束
      END
```

A. #0FFH, #4000H B. 0FFH, 4000H
C. #0FFH, #40FFH D. 0FFH, 40FFH

27. 编程, 将(30H),(31H)单元中的数与(40H),(41H)单元中的数相加, 结果存于(30H),(31H)单元中。

解:

```
MOV A, 30H
ADD A, 40H
MOV 30H, A
MOV A, 31H
ADDC A, 41H
MOV 31H, A
```


28. 编程, 根据累加器 A 中命令键键值, 设计命令键操作程序入口跳转表:

解:

```
CLR C ; 清进位
RLC A ; 键值乘 2
MOV DPTR, #JPTAB ; 指向命令键跳转表首址 0100H
JMP @A+DPTR ; 散转入命令键入口
JPTAB: AJMP CCS0 ; 双字节指令
      AJMP CCS1
      AJMP CCS2
```

提示: 从程序中看出, 当(A)=00H 时, 散转到 CCS0; 当(A)=01H 时, 散转到 CCS1;。由于 AJMP 是双字节指令, 散转前中的键值应乘 2。

29. 编程, 若片内 RAM 30H 单元内容大于 0, 40H 单元置 00; 30H 单元内容 < 0, 40H 单元置 FFH; 30H 单元内容等于 0, 40H 单元置 1。



答: CJNE 30H,#00H,LP1
 MOV 40H,#01H
 SJMP LP2
 LP1:JC LP3
 MOV 40H,#00H
 SJMP LP2
 LP3:MOV 40H,#0FFH
 LP2:SJMP \$

30.编程，数据块传送。将片内 RAM 20H 单元开始的 10 个字节，传送到片外 RAM 1000H 单元开始的单元中去。

解: ORG 0800H
 START: MOV R0, #20H
 MOV DPTR,#1000H
 MOV R7,#0AH
 LOOP: MOV A,@R0
 MOVX @DPTR,A
 INC R0
 INC DPTR
 DJNZ R7,LOOP
 SJMP \$

31.编程，片内 RAM30H 单元开始有 10 个字节的二进制数。编一子程序，求它们的和。（和 <256）将和存入 20H 单元中。

解: 方法 1

```

      ORG 1000H
ASUB1: MOV R0, #30H
      CLR A
      MOV R7, #10
      CLR C
LOOP:  ADDC A, @R0
      INC R0
      DJNZ R7, LOOP
      MOV 20H, A
      RET
  
```

方法 2

```
                ORG    2000H
ASUB2:  MOV    R0, #30H
        MOV    A, @R0
        MOV    R7, #09
        CLR    C
LOOP:   INC    R0
        ADDC   A, @R0
        DJNZ   R7, LOOP
        MOV    20H, A
        RET
```

32.编程，编写将 30H，31H 单元中的两字节二进制数乘以 2 的程序（积小于 65536）
解：

```
                ORG    1000H
MSUB:   CLR    C
        MOV    A, 30H
        RLC    A
        MOV    30H, A
        MOV    A, 31H
        RLC    A
        MOV    31H, A
        RET
```

第四章习题

1. 双字节加法程序段。设被加数存放于片内 RAM 的 addr1(低字节)和 addr2(高字节), 加数存放于片内 RAM 的 addr3(低字节)和 addr4(高字节), 结果放在 addr1(低字节)和 addr2(高字节)中。

解:

```
ORG 1000H
START: PUSH ACC;
      MOV R0, #addr1;
      MOV R1, #addr3;
      MOV A, @R0;
      ADD A, @R1;
      MOV @R0, A;
      INC R0;
      INC R1;
      MOV A, @R0;
      ADD A, @R1;
      MOV @R0, A;
      POP ACC;
```

2. 拆字。将片内 RAM 20H 单元的内容拆成两段, 每段四位。并将它们分别存入 21H 与 22H 单元中。程序如下:

解:

```
ORG 2000H
START: MOV R0, #21H ; 21H→R0
      MOV A, 20H ; (20H)→A
      ANL A, #0FH ; A∧#0FH→A
      MOV @R0, A ; A.→(R0)
      INC R0 ; R0+1→R0
      MOV A, 20H ; (20H)→A
      SWAP A ; A0~3←→A4~7
      ANL A, #0FH ; A∧#0FH→A
      MOV @R0, A ; A.→(R0)
      SJMP $
```

3.16 位数求补。设 16 位二进制数在 R1R0 中, 求补结果存于 R3R2 中。

解:

```
ORG 1000H
START: MOV A, R0
      CPL A
      ADD A, #01H
      MOV R2, A
      MOV A, R1
      CPL A
```

```

ADDC  A, #00H
MOV   R3, A
SJMP  $

```

4. 128 种分支转移程序。

功能：根据入口条件转移到 128 个目的地址。

入口：(R3)=转移目的地址的序号 00H~7FH。

出口：转移到相应子程序入口。

解：

```

JMP_128: MOV    A, R3
          RL     A
          MOV    DPTR, #JMPTAB
          JMP    @A+DPTR
JMPTAB:  AJMP    ROUT00
          AJMP    ROUT01
          ⋮
          AJMP    ROUT7F

```

} 128 个子程序首址

提示：此程序要求 128 个转移目的地址 (ROUT00 ~ROUT7FH) 必须驻留在与绝对转移指令 AJMP 相同的一个 2KB 存储区内。RL 指令对变址部分乘以 2，因为每条 AJMP 指令占两个字节。

5. 存放于 addr1 和 addr2 中的两个无符号二进制数，求其中的大数并存放于 addr3 中。程序如下：

```

解：      ORG    1000H
addr1  DATA  31H      ; 定义 addr1
addr2  DATA  32H      ; 定义 addr2
addr3  DATA  30H      ; 定义 addr3
START: MOV A, addr1
        CJNE A, addr2, LOOP1 ; 两数比较，不相等则转 LOOP1
        SJMP  LOOP2
LOOP1:  JC   LOOP2          ; 当 CY=1, 转 LOOP2
        MOV  addr3, A       ; CY=0, (A)>(addr2)
        SJMP LOOP3          ; 转结束
LOOP2:  MOV  addr3, addr2    ; CY=1, (addr2)>A.
LOOP3:  END

```

6. 片内 RAM ONE 和 TWO 两个单元中存有两个无符号数，将两个数中的小者存入 RES 单元。程序如下：

```

解：      ORG    1000H
ONE  DATA  22H ; 定义 ONE
TWO  DATA  23H ; 定义 TWO
RES  DATA  30H ; 定义 RES

```



```

MOV A, ONE
START:CJNE A, TWO, BIG
      SJMP STORE
BIG: JC  STORE
      MOV A, TWO
STORE: MOV RES, A
      SJMP $

```

7. 设变量 x 存放于 VAR 单元中，函数值 y 存放于 FUNC 中，按下式赋值

$$y = \begin{cases} 1 & x > 0 \\ 0 & x = 0 \\ -1 & x < 0 \end{cases}$$

解：程序如下：

```

ORG 0800H
VAR DATA 30H ;定义 VAR
FUNC DATA 31H ;定义 FUNC
START: MOV A, VAR ;取 x
      JZ COMP ;为 0 转 COMP
      JNB ACC.7, POSI ;x>0 转 POSI
      MOV A, #0FFH ;x<0,-1→A
      SJMP COMP
POSI: MOV A, #01H
COMP: MOV FUNC, A
      END

```

8. 以机器周期为单位，计算以下程序的执行时间。

```

MOV R7, #64H ;1 个机器周期
LOOP: MOV R6, #0FAH ;1
      DJNZ R6, $ ;(2), 250*2=500
      DJNZ R7, LOOP ;(2), 503*100=50300
      RET ;2

```

答：共 50303 个机器周期

9. 以机器周期为单位，计算以下程序的执行时间。

```

TIME: MOV R1, #0FAH ;1 个机器周期
L1: MOV R0, #0FFH ;1
W1: DJNZ R0, W1 ;(2), 255*2=510
      DJNZ R1, L1 ;(2), 513*250=128250
      NOP ;1
      NOP ;1
      RET ;2

```

答：共 128255 个机器周期

10. 1) 编写延时 4ms 的子程序。设晶振为 12MHz。

解：晶振为 12MHz，则 1 个机器周期为 1 微妙（1 μ s），一条 DJNZ 指令为 2 个机器周期，则执行该指令 2000 次为 4ms，2000=20*100。

```
ORG 1000H
Delay: MOV R6,#20
DLY1:  MOV R7,#100
DLY2:  DJNZ R7,DLY2
        DJNZ R6,DLY1
        RET
```

2) 编写延时 4ms 的子程序。设晶振为 6MHz。

解：晶振为 6MHz，则 1 个机器周期为 2 微妙（2 μ s），一条 DJNZ 指令为 2 个机器周期，则执行该指令 1000 次为 4ms，1000=20*50

```
ORG 1000H
Delay: MOV R6,#20
DLY1:  MOV R7,#50
DLY2:  DJNZ R7,DLY2
        DJNZ R6,DLY1
        RET
```

11. 多字节无符号加法程序。设被加数低字节地址存于 R0 中，加数低字节地址存于 R1 中，字节数存于 R3 中，相加结果存于原被加数单元。

解：

```
START: MOV A,R0          ;保存首地址
        MOV R5,A
        MOV A,R3          ;保存字节数
        MOV R7,A
        CLR C
ADDA:   MOV A,@R0
        ADDC A,@R1        ;作加法
        MOV @R0,A        ;部分和存入对应的被加数单元
        INC R0
        INC R1
        DJNZ R7,ADDA      ;当(R7) 不等于 0 时，则继续作加法
        JNC AADB          ;处理最高字节进位
        INC R3
        MOV @R0,#01H
ADDB:   MOV A,R5          ;增加内容
        MOV R0,A
        END
```

12. 搜索最大值。从片内 RAM 的 BLOCK 单元开始有一个无符号数据块，其长度存于 LEN 单元中，试求出其中最大的。

解：

```

START:  LEN DATA 20H
        MAX DATA 21H
        BLOCK DATA 22H
        CLR A
        MOV R2, LEN      ;数据块长度送 R2
        MOV R1, # BLOCK  ;置地址指针
LOOP:   CLR C
        SUBB A, @ R1     ;用减法做比较
        JNC NEXT        ;无借位 A 大
        MOV A, @ R1     ;否则大者送 A
        SJMP NEXT1
NEXT:   ADD A, @ R1      ;A 大恢复 A
NEXT1:  INC R1           ;修改地址指针
        DJNZ R2, LOOP    ;未完继续
        MOV MAX, A       ;若完则存大数

```

13. 设被加数存放在内部 RAM 的 20H、21H 单元，加数存放在 22H、23H 单元，若要求和存放在 24H、25H 中，试编写出 16 位无符号数相加的程序（采用大端模式存储）。

答：程序如下：

```

ORG    0000H
MOV     R0, #21H
MOV     R1, #23H
MOV     A, @R0
ADD     A, @R1
MOV     25H, A
DEC     R0
DEC     R1
MOV     A, @R0
ADDC    A, @R1
MOV     24H, A
SJMP    $
END

```

14. 计算下面子程序中指令的偏移量和程序执行的时间（晶振频率为 12MHz）。

```

MOV     R3, #15H      ;1 个机器周期
DL1:MOV  R4, #255      ;1 个机器周期
DL2:MOV  P1, R3        ;2 个机器周期
        DJNZ R4, DL2   ;2 个机器周期
        DJNZ R3, DL1   ;2 个机器周期
        RET           ;2 个机器周期

```

答：15348us

分析： $((2+2) \times 255 + 1 + 2) \times 15 + 1 + 2 = 15348\text{us}$

15. 试编写程序，查找在内部 RAM 的 30H~50H 单元中是否有 0AAH 这一数据。若有，则

将 51H 单元置为“01H”；若未找到，则将 51H 单元置为“00H”。

```
答：START:    MOV     R0,#30H
              MOV     R2,#20H
              LOOP:   MOV     A,@R0
                   CJNE    A,#0AAH,NEXT
                   MOV     51H,#01H
                   LJMP    EXIT
              NEXT:   INC     R0
                   DJNZ    R2,LOOP
                   MOV     51H,#00H
              EXIT:   RET
```

第五章习题

1. 假设允许片内定时器/计数器中断，禁止其他中断。试根据假设条件设置 IE 的相应值。

解：(IE) = 10001010B = 8AH

(a) 用字节操作指令

MOV IE, #8AH; 或 MOV A8H, #8AH;

(b) 用位操作指令

SETB ET0; 定时器/计数器 0 允许中断

SETB ET1; 定时器/计数器 1 允许中断

SETB EA ; CPU 开中断

2. 设 80C51 的片外中断为高优先级，片内中断为低优先级。设置 IP 相应值。

解：(IP) = 00000101B = 05H

(a) 用字节操作指令

MOV IP, #05H; 或 MOV 0B8H, #05H;

(b) 用为操作指令

SETB PX0


SETB PX1

CLR PS

CLR PT0

CLR PT1

3. 外部中断实验：如图所示，将 P1 口的 P1.4~P1.7 作为输入位，P1.0~P1.3 作为输出位。要求利用 89C51 将开关所设的数据读入单片机内，并依次通过 P1.0~P1.3 输出，驱动发光二极管，以检查 P1.4~P1.7 输入的电平情况（若输入为低电平则相应的 LED 亮）。现要求采用中断边沿触发方式，每中断一次，完成一次读/写操作。



解：如图所示，采用外部中断 0，中断申请从 $\overline{\text{INT0}}$ 输入，并采用了去抖动电路。

当 P1.0~P1.3 的任何一位输出为 0 时，相应的发光二极管就会发光。当开关 S1 闭合时，发出中断请求。中断服务程序的矢量地址为 0003H。源程序如下。

源程序如下：


```
ORG    0000H
LJMP   MAIN    ; 上电，转向主程序
ORG    0003H   ; 外部中断 0 入口地址
LJMP   INSERT  ; 转向中断服务程序
```

```

    ORG    0030H          ; 主程序
        MOV    SP    #60H
MAIN:  SETB    EX0        ; 允许外部中断 0 中断
        SETB    IT0      ; 选择边沿触发方式
        SETB    EA       ; CPU 开中断
HERE:  SJMP    HERE      ; 等待中断
ORG    0200H            ; 中断服务程序
INSER: MOV    A, #0F0H
        MOV    P1, A     ; 设 P1.4~P1.7 为输入
        MOV    A, P1     ; 取开关数
        SWAP   A         ; A 的高、低四位互换
        MOV    P1, A     ; 输出驱动 LED 发光
        RETI            ; 中断返回
END

```

4.利用中断显示系统故障:如图所示,此中断电路可实现系统的故障显示。当系统的各部分正常工作时,四个故障源的输入均为低电平,显示灯全不亮。当有某个部分出现故障时,则相应的输入线由低电平变为高电平,相应的发光二极管亮。



解: 如图所示, 当某一个故障信号输入线由低电平变为高点平时, 会通过 $\overline{\text{INT0}}$ 线引起 89C51 中断(边沿触发方式)。在中断服务程序中, 应将各故障源的信号读入, 并加以查询, 以进行相应的发光显示。

源程序如下:

```

    ORG    0000H
    LJMP   MAIN      ; 上电, 转向主程序
    ORG    0003H     ; 外部中断 0 入口地址
    LJMP   INSER     ; 转向中断服务程序
    MOV    SP    #60H
MAIN:  ANL    P1, #55H

```

```

                                ; P1.0,P1.2,P1.4,P1.6 为输入
                                ; P1.1,P1.3,P1.5,P1.7 为输出
                                SETB  EX0  ; 允许外部中断 0 中断
                                SETB  IT0  ; 选择边沿触发方式
                                SETB  EA   ; CPU 开中断
HERE:    SJMP  HERE    ; 等待中断
INSER:   JNB   P1.0, L1 ; 查询中断源, ( P1.0) =0, 转 L1
                                SETB  P1.1 ; 是 P1.0 引起的中断, 使相应的二极管亮
L1:      JNB   P1.2, L2 ; 继续查询
                                SETB  P1.3 ;
L2:      JNB   P1.4, L3
                                SETB  P1.5
L3:      JNB   P1.6, L4;
                                SETB  P1.7
L4:      RETI
                                END

```

5. 某系统有三个外部中断源 1、2、3，当某一中断源变低电平时便要求 CPU 处理，它们的优先处理次序由高到低为 3、2、1，处理程序的入口地址分别为 2000H、2100H、2200H。试编写主程序及中断服务程序（转至相应的入口即可）。

答：将 3 个中断信号经电阻线接 $\overline{\text{INT1}}$ 。

```

                                ORG  0000H
                                LJMP  MAIN
                                ORG  00013H
                                LJMP  ZDFZ
                                ORG  0040H
MAIN:   SETB  EA
                                SETB  EX1
                                SJMP  $
                                ORG  0200H
ZDFZ:  PUSH  PSW
                                PUSH  ACC
                                JB     P1.0,DV0
                                JB     P1.1,DV1
                                JB     P1.2,DV2
INRET:  POP   ACC
                                POP   PSW
                                RETI
                                ORG  2000H
DV0:-----
                                JMP    INRET
                                ORG  2100H
DV1:-----
                                JMP    INRET

```

```

        ORG    2200H
DV2:-----
        JMP    INRET

```

6. 以中断方法设计 51 单片机秒、分脉冲发生器。假定 P1.0 每秒产生一个机器周期的正脉冲，P1.1 每分钟产生一个机器周期的正脉冲，晶振频率为 6MHz。

答：（1）实现时钟计时的基本方法

1) 计算计数初值。

时钟计时的最小单位是秒，可把定时器的定时时间定为 50ms，计数溢出 20 次即得到 1 秒；20 次计数可用软件方法实现。

假定使用定时器 T0，以工作模式 1 进行 50ms 的定时。如 fosc=6MHz，则计数初值 X 为：

∴ X=15536=3CB0H

因此：(TL0)=0B0H (TH0)=3CH

2) 采用定时方式进行溢出次数的累计，计满 20 次即得到秒计时。

设置软件计数器初值为 20，每 50ms 定时时间到溢出中断，使软件计数器减 1，直到减到 0，则 1s 到。

3) 设置累加单元对 1s 进行计数，，计满 60 次即得到分计时。

（2）到了这些时刻，分别在 P1.0、P1.1 输出正脉冲即可。

```

ORG    0000H
AJMP   START
ORG    000BH           ; 定时器 0 中断入口.
AJMP   SERVE

START:
MOV    TMOD, #01H      ;T0 定时方式 0
MOV    TH0, #3CH       ;定时
MOV    TL0, #0B0H
SETB   TR0             ;启动.
SETB   ET0             ;允许中断.
SETB   EA
MOV    R2, #20          ;秒.
MOV    R3, #60          ;分.
CLR    P1.0            ;开始时，输出低电平.
CLR    P1.1
SJMP   $               ;等待中断.

SERVE:                  ;50ms 执行一次本中断程序.
MOV    TL0, #0B0H      ;重新送入初始值.
MOV    TH0, #3CH       ;定时
DJNZ   R2, T0_END      ;不到 20 次转移.
SETB   P1.0            ;到了 20 次，就是到了 1 秒，输出正脉冲.
CLR    P1.0            ;马上就恢复为 0，这就是 1 个机器周期.
MOV    R2, #20
DJNZ   R3, T0_END      ;不到 60 秒转移.
SETB   P1.1            ;到了 60 秒，输出正脉冲.

```




```

CLR P1.1          ;马上就恢复为 0，这就是 1 个机器周期.
MOV R3, #60
T0_END:
    RETI
END

```

7. 外部数据经 P1 口输入 89C51 单片机，每准备好一个数据，外设发出选通信号，经触发器和非门至 INT0。通过 P3.0 口输出低电平将中断请求撤除，并将数据传送到片外 RAM 1000H 单元开始的单元中去。



答：数据传送过程：选通信号有效，发中断请求；执行中断读取数据，同时将中断请求撤除（P3.0=0，使 INT0=1）。

```

ORG 0000H
START: LJMP MAIN
ORG 0003H
LJMP IINT0
ORG 0030H          ;主程序
MAIN: CLR IT0      ;低电平触发
    SETB EA
    SETB EX0        ;开中断
    MOV DPTR, #1000H
    ...
IINT0: PUSH PSW     ;中断处理程序
    PUSH ACC
    CLR P3.0
    NOP
    NOP
    SETB P3.0        ;撤除 INT0
    MOV A, P1         ;输入数据
    MOVX @DPTR, A     ;保存数据

```

INC DPTR

...

POP ACC ;恢复现场

POP PSW

RETI ;返回主程序

第六章习题

1.用 89C51 单片机的定时器 T1 作计数方式，用模式 2，则工作方式控制字为（ A ）。

A.60H B.02H C.06H D.20H

用 89C51 单片机的定时器 T1 作定时方式，用模式 1，则初始化编程为（ C ）。

A. MOV TMOD, #01H B.MOV TMOD, #50H
B. MOV TMOD, #10H D.MOV TCON, #02H

用 8031 的定时器 T1 作定时方式，用模式 2，则工作方式控制字为（ D ）。

A.60H B.02H C.06H D.20H

用 89C51 单片机的定时器 T1 作定时方式，用模式 2，则初始化编程为（ B ）。

A. MOV TMOD, #06H
B. MOV TMOD, #20H
C. MOV TMOD, #10H
D. MOV TMOD, #60H

89C51 单片机启动定时器 0 开始计数的指令是使 TCON 的（ B ）。

A. TF0 位置 1 B.TR0 位置 1 C.TR0 位置 0 D.TR1 位置 0

2.定时器/计时器 T0 的初始化程序如下：

```
MOV TMOD, #06H
MOV TH0, #0FFH
MOV TLO, #0FFH
SETB EA
SETB ET0
```

执行该程序段后，把定时器/计时器 T0 的工作状态设置为（ D ）。

A. 工作方式 0，定时应用，定时时间 2 us，中断禁止
B.工作方式 1，计数应用，计数值 255，中断允许
C.工作方式 2，定时应用，定时时间 510 us，中断禁止
D.工作方式 2，计数应用，计数值 1，中断允许

3.某 89C51 单片机系统的晶振频率为 11.0592MHz，使用 T0（或 T1）以模式 1 定时，则最大定时时间为（ D ）ms。

A.0.278; B.8.889; C.71.11; D.71.111

4. 在 89C51 单片机系统中，设定定时器 T0 用于定时 10ms，晶振频率为 6MHz。试确定 T0 初值。并编写定时器 T0 初始化程序段。

解：1) 确定 T0 初值

当 T0 处于工作模式 1 时，加 1 计数器为 16 位。

定时时间：

$$T = (2^{16} - T0\text{初值}) \times \text{振荡周期} \times 12$$

所以

$$10ms = (2^{16} - T0\text{初值}) \times \text{振荡周期} \times 12$$

$$T0\text{初值} = 2^{16} - \frac{10 \times 10^{-3} \times 6 \times 10^6}{12}$$

$$\therefore T0\text{初值} = 60536 = 0EC78H$$

即：(TH0) = ECH (高 8 位)

(TL0) = 78H (低 8 位)

2) 确定 TMOD 的初值



TMOD 的低四位为 T0 的控制位

模式 0: M1M0=01,

定时器方式: C/T=0,

门控位不受 INT0 的影响: GATE=0,

其余位: 为 0。

$$\therefore \text{模式字为 TMOD} = 0000\ 0001\ B = 01H$$

3) 编写定时器 T0 的初始化程序段

主程序:

```
ORG    0100H
MAIN:  MOV    SP,#60H      ; 设堆栈指针
        MOV    TMOD,#01H   ; 选择工作模式
        MOV    TH0,#ECH    ; 送初值
        MOV    TL0,#78H
        SETB   TR0         ; 启动定时
        ...
```

5. 在 89C51 单片机系统中, 设定时器 T0 选择工作模式 1 的计数器工作方式, 其计数器初值为 FFFFH, 问此时定时器 T0 的实际用途是什么?

解: 因其初值为 FFFFH, 只要随机外来一脉冲即可溢出, 向 CPU 申请中断, 故这一内部中断源实质上已作为外部中断源使用。因此此定时器 T0 并不用于定时或计数。

6. 在 89C51 单片机系统中, 设晶振频率为 12MHz, 试计算定时器 T0 工作于模式 1 时的最大定时时间 T。

解: 当 T0 处于工作模式 1 时, 加 1 计数器为 16 位。

$$\text{定时时间为: } t = (2^{16} - T0\text{初值}) \times \text{振荡周期} \times 12$$


最大定时时间为“T0 初值=0”时。

所以:

$$T = 2^{16} \times \text{振荡周期} \times 12$$

$$\begin{aligned}
&= 2^{16} \times \frac{1}{12 \times 10^6} \times 12 \\
&= 65536 \times 10^{-6} \\
&= 65.536 \text{ ms}
\end{aligned}$$

7.在 89C51 单片机系统中，用定时器 T1 产生一个 50Hz 的方波，由 P1.1 输出。使用程序查询方式，晶振频率为 12MHz。



解：1) 确定定时器 T1 初值

- ∵ 方波周期 $T = 1/50 = 0.02\text{s} = 20\text{ms}$
- ∴ 用 T1 定时 10ms, 时间到 P1.1 引脚电平取反。

定时时间: $t = (2^{16} - T1\text{初值}) \times \text{振荡周期} \times 12$

$$\begin{aligned}
T1\text{初值} &= 2^{16} - \frac{t}{\text{振荡周期} \times 12} \\
\therefore T1\text{初值} &= 2^{16} - \frac{10 \text{ ms}}{\frac{1}{6 \times 10^6} \times 12}
\end{aligned}$$

$$\therefore T1\text{初值} = 50536 = \text{D8F0H}$$

$$\therefore (\text{TH1}) = \text{D8H}, (\text{TL1}) = \text{F0H}$$

2) 确定工作模式寄存器 TMOD 的值

- ∵ 定时器 T1 工作于模式 1 的定时器工作方式,
- ∴ 高四位: GATE=0, C/T=0, M1M0=01, 低四位: 取 0。
- ∴ (TMOD) = 0001 0000 B = 10H

3) 编程

```

ORG    0000H
LJMP   START
ORG    0080H
START: MOV    TMOD,#10H    ; T1 为模式 1
      MOV    TL1,#F0H      ; 送初值
      MOV    TH1,#0D8H
      SETB   P1.1          ; P1.1 置 1
      SETB   TR1           ; 启动定时
LOOP:  JNB    TF1, LOOP    ; 查询定时时间到否
      CLR    TF1           ; 产生溢出, 清标志位
      MOV    TL1,#F0H      ; 重新置初值
      MOV    TH1,#0D8H


```

```

CPL    P1.1        ; 取反
SJMP   LOOP        ; 重复循环

```

8.在 89C51 单片机系统中,当 P3.4 引脚(定时器 T0 的外部计数输入端)上的电平发生负跳变时,从 P1.0 输出一个 500μs 的同步脉冲。用查询方式编程实现该功能,晶振频率为 6MHz。



解:

(1) 确定 TMOD 值

门控位 GATE=0, 高四位取 0, 选 T0 为模式 2 (M1M0=10),

首先为计数器工作方式 (C/T=1): (TMOD)=0000 0110B=06H

当 P3.4 引脚上的电平发生负跳变时, T0 计数器加 1, 溢出标志 TF0 置 1;

然后改变 T0 为 500μs 定时工作方式 (C/T=0): (TMOD)=02H

当上面 TF0=1 时使 P1.0 输出由 1 变为 0。开始 T0 定时, 到 500μs 产生溢出, 使 P1.0 输出恢复高电平, T0 又恢复外部事件计数方式。

(2) 计算 T0 初值

T0 工作在外部事件计数方式:

题目要求 T0 引脚出现一次外部事件时, 引起 T0 中断。

所以: 设计计数器初值为 FFH, 当计数器再加 1 就溢出了。

即: T0 初值+1=2⁸

T0 初值 = 2⁸ - 1 = 11111111B = 0FFH

即: (TH0) = (TL0) = 0FFH

T0 工作在定时工作方式:

因为晶振频率为 6MHz, 定时 500μs。

所以: (2⁸ - T0 初值) × 2μs = 500μs

得出: T0 初值 = 6 = 06H,

即: (TH0) = (TL0) = 06H

(3) 程序清单:

```

                ORG    0000H
                LJMP   START
                ORG    0080H
START:          MOV    TMOD,#06H        ; 设置 T0 为模式 2,外部计数方式
                MOV    TL0,#0FFH        ; T0 计数器初值
                MOV    TH0,#0FFH
                SETB   TR0                ; 启动 T0 计数
LOOP1:          JBC    TF0, PTF01        ; 查询 T0 溢出标志, TF0=1 时转, 且清 TF0=0
                SJMP   LOOP1              ; 等待 T0 溢出
PTF01:          CLR    TR0                ; 停止计数
                MOV    TMOD,#02H        ; 设置 T0 为模式 2, 定时方式
                MOV    TL0,#06H          ; 送初值, 定时 500μs
                MOV    TH0,#06H

```

```

                CLR    P1.0                ; P1.0 清 0
                SETB   TR0                ; 启动定时 500μs
LOOP2:          JBC    TF0,PTFO2          ; 查询 T0 溢出标志, TF0=1 时转并清 TF0=0
                SJMP   LOOP2              ; 等待 T0 溢出中断 (定时时间到)
PTFO2:          SETB   P1.0              ; P1.0 置 1
                CLR    TR0                ; 停止定时
                SJMP   START              ; 转向开始, 重新等待 T0 引脚的脉冲

```

9.在 89C51 单片机系统中, 利用定时器 T1 的模式 2 对外部信号计数。要求每计满 100 次, 将 P1.0 端取反。

解:

(1) 选择模式

外部信号由 T1 (P3.5) 引脚输入, 每发生一次负跳变计数器加 1, 每输入 100 个脉冲, 计数器发生溢出中断, 中断服务程序将 P1.0 取反一次。

T1 计数工作方式模式 2 的模式字为 (TMOD)=60H。

T0 不用时, TMOD 的低 4 位可任取, 但不能进入模式 3, 一般取 0。

(2) 计算 T1 的计数初值

$$X = 2^8 - 100 = 156 = 9CH$$

因此:TL1 的初值为 9CH, 重装初值寄存器 TH1=9CH。

(3) 程序清单

```

                ORG     0000H
                LJMP    MAIN
                ORG     001BH              ; 中断服务程序入口
                CPL     P1.0
                RETI
MAIN:           MOV     TMOD,#60H         ; 设置 T1 为模式 2,外部计数方式
                MOV     TL1,#9CH          ; T1 计数器初值
                MOV     TH1,#9CH
                MOV     IE, #88H          ; 定时器开中断
                SETB    TR1                ; 启动 T1 计数
HERE:           SJMP    HERE              ; 等待中断

```

10.在 89C51 单片机系统中, 设某用户系统已使用了两个外部中断源, 并置定时器 T1 工作在模式 2, 作串行口波特率发生器用。现要求再增加一个外部中断源, 并由 P1.0 引脚输出一个 5kHz 的方波, 晶振频率为 12MHz。

解: (1) 工作模式

可设置 T0 工作在模式 3 计数器方式, 把 T0 的引脚作附加的外部中断输入端, TL0 的计数初值为 FFH, 当检测到 T0 引脚电平出现由 1 至 0 的负跳变时, TL0 产生溢出, 申请中断。

T0 模式 3 下, TL0 作计数用, 而 TH0 用作 8 位的定时器, 定时控制 P1.0 引脚输出 5kHz 的方波信号。

(2) 计算初值

TL0 的计数初值为 FFH。

TH0 的计数初值 X 为: P1.0 方波周期 $T = 1/(5kHz) = 0.2ms = 200\mu s$

用 TH0 作定时 100μs 时, $X = 256 - 100 \times 12/12 = 156$

(3) 程序清单

```
MOV    TMOD,#27H    ; T0 为模式 3,计数方式, T1 为模式 2, 定时方式
MOV    TL0,#0FFH    ; TL0 计数初值
MOV    TH0,#156      ; TH0 计数初值
MOV    TL1,#data     ; data 是根据波特率, 要求设置的常数(初值)
MOV    TH1,#data
MOV    TCON,#55H     ; 外中断 0, 外中断 1 边沿触发, 启动 T0, T1
MOV    IE, #9FH      ; 开放全部中断
TL0INT: MOV    TL0, #0FFH ; TL0 重赋初值
                                ; (中断处理)
RET
                                ;TH0 溢出中断服务程序(由 001BH 转来)
TH0INT: MOV    TH0, #156 ; TH0 重赋初值
CPL    P1.0          ;P1.0 取反输出
RET
                                ;串行口及外部中断 0, 外部中断 1 的服务程序略
```

11. 在 89C51 单片机系统中, 设时钟频率为 6MHz, 编写利用 T0 产生 1s 定时的程序。

解: (1) 定时器 T0 工作模式的确定

∵ 模式 0 最长可定时 16.384ms; 模式 1 最长可定时 131.072ms;

模式 2 最长可定时 512μs;

∴ 定时 1s, 可选用模式 1, 每隔 100ms 中断一次, 中断 10 次从而达到 1s 的定时。

(2) 求计数器初值 X

∵ $(2^{16}-X) \times 12 / (6 \times 10^6) = 100 \times 10^{-3} \text{ s}$

∴ $X = 15536 = 3CB0H$

因此: $(TL0) = 0B0H$, $(TH0) = 3CH$

(3) 实现方法: 对于中断 10 次计数, 可使 T0 工作在计数方式, 也可用循环程序的方法实现。本例采用循环程序法。

(4) 源程序清单

```
ORG    0000H
AJMP   MAIN      ; 上电, 转向主程序
ORG    000BH     ; T0 的中断服务程序入口地址
AJMP   SERVE     ; 转向中断服务程序
ORG    0080H     ; 主程序
MAIN:  MOV    SP,#60H ; 设堆栈指针
MOV    B, #0AH   ; 设循环次数
MOV    TMOD,#01H ; 设置 T0 工作于模式 1
MOV    TL0,#0B0H ; 装计数值低 8 位
MOV    TH0,#3CH  ; 装计数值高 8 位
SETB   TR0       ; 启动定时
SETB   ET0       ; T0 开中断
SETB   EA        ; CPU 开中断
SJMP   $         ; 等待中断
SERVE: MOV    TL0,#0B0H ; 重新赋初值
```




```

        MOV    TH0,#3CH
        DJNZ   B,LOOP      ; B-1 不为 0，继续定时
        CLR    TR0         ; 1s 定时到，停止 T0 工作
LOOP:    RETI              ; 中断返回
        END

```

12. 利用 89C51 单片机测量 INT0 引脚上出现的正脉冲宽度，并将结果（以机器周期的形式）存放在 70H 和 71H 两个单元中。



解：将 T0 设置为方式 1 的定时方式，且 GATE=1，计数器初值为 0，将 TR0 置 1。INT0 引脚上出现高电平时，加 1 计数器开始对机器周期计数。 $\overline{\text{INT0}}$ 引脚上信号变为低电平时，停止计数。然后读出 TH0、TL0 的值

程序如下：

```

        ORG    0000H
        AJMP   MAIN
        ORG    0200H
MAIN:    MOV    TMOD,#09H    ;置 T0 为定时器方式 1，GATE=1
        MOV    TH0,#00H     ;置计数初值
        MOV    TL0,#00H
        MOV    R0,#71H      ;置地址指针初值（指向低字节）
L1:      JB     P3.2,L1      ;高电平等待
        SETB   TR0          ;当  $\overline{\text{INT0}}$  由高变低时使 TR0=1，准备好
L2:      JNB    P3.2,L2      ;等待  $\overline{\text{INT0}}$  变高
L3:      JB     P3.2,L3      ;已变高，启动定时，直到  $\overline{\text{INT0}}$  变低
        CLR    TR0          ; $\overline{\text{INT0}}$  由高变低，停止定时
        MOV    @R0,TL0      ;存结果
        INC    R0
        MOV    @R0,TH0
        SJMP   $
        END

```

13. 已知 89C51 单片机时钟频率为 6MHz，请利用 T0 和 P1.0 输出矩形波。矩形波高电平宽 50 μs ，低电平宽 300 μs 。

解：T0 采用模式 2 作为 50 μs 定时时的初值：(TL0)=0E7H (TH0)=0E7H

T0 采用模式 2 作为 300 μs 定时时的初值：(TL0)=06AH (TH0)=06AH

```

        MOV    TMOD,#02H    ;设置定时器 T0 工作于模式 2
L2:    CLR    P1.0          ;P1.0 输出低电平
        MOV    TH0,#6AH     ;置入定时 300 $\mu\text{s}$  初值
        MOV    TL0,#6AH
        SETB   TR0          ;启动 T0

```

```

LOOP1: JBC  TF0,L1      ;查询300μs 时间到? 时间到, 转 L1
        SJMP  LOOP1     ;时间未到, 转 LOOP1, 继续查询
L1:     SETB  P1.0       ;P1.0输出高电平
        CLR   TR0        ;关闭 T0
        MOV  TH0,#0E7H   ;置入定时50μs 初值
        MOV  TL0,# 0E7H
        SETB  TR0        ;启动 T0
LOOP2: JBC  TF0,L2      ;查询50μs 时间到? 时间到, 转 L2
        SJMP  LOOP2     ;时间未到, 转 LOOP2, 继续查询

```

14. 已知 89C51 单片机的时钟频率为 12MHz, 用 T1 定时。试编程由 P1.0 和 P1.1 引脚分别输出周期为 2ms 和 500μs 的方波。

解: 采用模式2作为250μs 定时初值: (TL1)=06H (TH1)=0F8H

计满4次即得到1ms

```

        MOV  R2,#04H     ;R2为“250μs”计数器, 置入初值4
        CLR  P1.0        ;P1.0输出低电平
        CLR  P1.1        ;P1.1输出低电平
        MOV  TMOD,#00H
L2:     MOV  TH1,#0F8H    ;置入定时250μs 初值
        MOV  TL1,#06H
        SETB  TR1        ;启动 T1
LOOP:   JBC  TF1,L1      ;查询250μs 时间到? 时间到, 转 L1
        SJMP  LOOP       ;时间未到, 转 LOOP, 继续查询
L1:     CPL  P1.1        ;P1.1输出取反, 形成周期为500μs
        CLR  TR1         ;关闭 T1
        DJNZ R2,L2       ;“250μs”计数器减1, 到1ms 吗? 未到转 L2
        CPL  P1.0        ;P1.0输出取反, 形成周期为2ms 方波
        MOV  R2,#04H     ;重置“250μs”计数器初值4
LJMP    L2               ;重复循环

```

15.89C51 单片机系统中, 设时钟频率为 12MHz。试编制一段程序, 功能为: 对定时器 T0 初始化, 使之工作在模式 2, 产生 200μs 定时, 并用查询 T0 溢出标志的方法, 控制 P1.0 输出周期为 2ms 的方波。

解: T0采用模式2作为200μs 定时初值: (TL1)=038H (TH1)=038H

计满5次即得到1ms

程序1:

```

        CLR  P1.0        ;P1.0输出低电平
        MOV  R2,#05H     ;R2为“200μs”计数器, 置入初值5, 计1ms
        MOV  TMOD,#02H   ;设定时器 T0工作于模式2
L2:     MOV  TH0,#38H     ;置入定时初值
        MOV  TL0,#38H
        SETB  TR0        ;启动 T0
LOOP:   JBC  TF0,L1      ;查询200μs 时间到? 时间到, 转 L1
        SJMP  LOOP       ;时间未到, 转 LOOP, 继续查询

```

```

L1:CLR  TR0          ;关闭 T0
      DJNZ R2,L2     ;“200μs”计数器减1，到1ms 吗？未到，转 L2
      CPL  P1.0      ;到1ms，P1.0取反，形成周期为2ms 的方波
      MOV  R2,#05H   ;重置“200μs”计数器初值
      LJMP L2        ;重复循环

```

程序2:

```

MAIN:  MOV TMOD,#02H ;设定时器 T0工作于模式2
      CLR  P1.0      ;P1.0输出低电平
LOOP1: MOV R2,#05H   ;R2为“200μs”计数器，置入初值5，计1ms
LOOP:  MOV TH0,#38H  ;置入定时初值
      MOV TL0,#38H
      SETB TR0       ;启动 T0
      JNB TF0,$      ;查询200μs 时间到？时间到，继续查询
      CLR  TR0       ;关闭 T0
      DJNZ R2,LOOP   ;“200μs”计数器减1，到1ms 吗？未到，转 LOOP
      CPL  P1.0      ;到1ms，P1.0取反，形成周期为2ms 的方波
      SJMP LOOP1     ;重复循环

```

16. 89C51 单片机系统中用一个定时/计数器加软件计数器的方式，实现一秒的时钟基准信号，试写出程序。（设晶振频率为 12MHz，由 P1.0 口输出秒信号。）

解:

本程序使用定时器 T0，工作模式 1。由于晶振频率为 12MHz，因此利用定时器 T0 计时 50ms，其初值为 3CB0H。利用工作寄存器 R7 作软件计数器，计数 20 次。每计时满一秒，就将 P1.0 口输出信号取反，以输出秒信号。

```

      ORG      0000H
      LJMP     MAIN
      ORG      000BH
      LJMP     ONE
      ORG      0030H
MAIN:  MOV     P1,0FFH
      MOV     SP,#60H
      MOV     R7,#20
      MOV     TMOD,#01H
      MOV     TL0,#0B0H
      MOV     TH0,#3CH
      SETB    TR0
      SETB    ET0
      SETB    EA
      SJMP     $
ONE:   PUSH    PSW
      PUSH    ACC
      MOV     TL0,#0B0H
      MOV     TH0,#3CH
      DJNZ    R7,LOOP

```

```
        CPL        P1.0
        MOV        R7,#20
LOOP:   POP        ACC
        POP        PSW
        RETI
        END
```

第七章习题

1.89C51 单片机系统中设串行口工作于方式 1，晶振频率为 11.059MHz，波特率为 9600 位/秒，SMOD=0，定时器 T1 工作模式 2，则定时器 1 的计数初值为(C)。

A.F8H B.FAH C.FDH D.FFH

分析:

$$X = 256 - \frac{11.059 \times 10^6 \times (0+1)}{32 \times 12 \times 9600} \cong 253 = 0FDH$$

2.89C51 单片机系统中设串行口工作于方式 1，晶振频率为 6MHz，波特率为 1200 位/秒，SMOD=0，定时器 T1 工作模式 2，则定时器 1 的计数初值为(C)。

A. FIH B.F4H C.F3H D.FOH

3.设串行口工作于方式 1，晶振频率为 6MHz，波特率为 2400 位/秒，SMOD=1，定时器 T1 工作模式 2，则定时器 1 的计数初值为(C)。

A.FIH B.F4H C.F3H D.FOH

4.89C51 单片机中若设串行控制寄存器 SCON=40H，则串行口的工作方式是(B)。

A.方式 0 B.方式 1 C.方式 2 D.方式 3

5.89C51 单片机中已知 f_{osc} 、SMOD 和波特率，试求串行方式 1 和 T1 定时初值。

① $f_{osc}=12\text{MHz}$ ，SMOD=0，bit/s=2400；

② $f_{osc}=6\text{MHz}$ ，SMOD=1，bit/s=1200；

③ $f_{osc}=11.0592\text{MHz}$ ，SMOD=1，bit/s=9600；

④ $f_{osc}=11.0592\text{MHz}$ ，SMOD=0，bit/s=2400；

解:

① $f_{osc}=12\text{MHz}$ ，SMOD=0，bit/s=2400；

$$T1_{\text{初值}} = 256 - (2^{\text{SMOD}}/32) * f_{osc} / (12 * \text{波特率}) \approx 256 - 13.02 = 243 = F3H$$

② $f_{osc}=6\text{MHz}$ ，SMOD=1，bit/s=1200；

$$T1_{\text{初值}} = 256 - (2^{\text{SMOD}}/32) * f_{osc} / (12 * \text{波特率}) \approx 256 - 26.04 = 230 = E6H$$

③ $f_{osc}=11.0592\text{MHz}$ ，SMOD=1，bit/s=9600；

$$T1_{\text{初值}} = 256 - (2^{\text{SMOD}}/32) * f_{osc} / (12 * \text{波特率}) \approx 256 - 3 = 253 = FAH$$

④ $f_{osc}=11.0592\text{MHz}$ ，SMOD=0，bit/s=2400；

$$T1_{\text{初值}} = 256 - (2^{\text{SMOD}}/32) * f_{osc} / (12 * \text{波特率}) \approx 256 - 12 = 244 = F4H$$

6. 89C51 单片机串行口工作在方式 1 和方式 3 时，其波特率与 f_{osc} 、定时器 T1 工作模式 2 的初值及 SMOD 位的关系如何？设 $f_{osc}=6\text{MHz}$ ，现利用定时器 T1 模式 2 产生的波特率为 110b/s。试计算定时器初值。

解:

关系如下:

$$X = 256 - \frac{11.059 \times 10^6 \times (\text{SMOD} + 1)}{32 \times 12 \times \text{波特率}}$$

当波特率为110b/s, $f_{\text{osc}}=6\text{MHz}$, 令 $\text{SMOD}=0$, 有: $X=114=072\text{H}$

7.89C51 单片机时钟振荡频率为 11.0592MHz, 选用定时器 T1 工作模式 2 作为波特率发生器, 波特率为 2400b/s, 求初值。

解: 设置波特率控制为 $(\text{SMOD})=0$

$$X \approx 256 - \frac{11.0592 \times 10^6 \times (0 + 1)}{384 \times 2400} = 244 = \text{F4H}$$

所以, $(\text{TH1})=(\text{TL1})=\text{F4H}$ 。

8.89C51 单片机串行口按双工方式收发 ASCII 字符, 最高位用来作奇偶校验位, 采用奇校验方式, 要求传送的波特率为 1200b/s, 时钟振荡频率为 6MHz, 设发送数据区的首地址为 20H, 接收数据区的首地址为 40H。编写有关的通信程序。

解: 7 位 ASCII 码加 1 位奇校验共 8 位数据, 故可采用串行口方式 1。

89C51 单片机的奇偶校验位 P 是当累加器 A 中 1 的数目为奇数时, $P=1$ 。如果直接把 P 的值放入 ASCII 码的最高位, 恰好成了偶校验, 与要求不符。因此, 要把 P 的值取反以后放入 ASCII 码最高位, 才是要求的奇校验。

发送和接收都通过调用子程序来完成, 设发送数据区的首地址为 20H, 接收数据区的首地址为 40H, f_{osc} 为 6MHz, 通过查波特率初值 (表 7-2) 可知定时器的初装值为 F3H。定时器 T1 采用工作模式 2, 可以避免计数溢出后用软件重装定时初值的工作。

$$X \approx 256 - \frac{6 \times 10^6 \times (0 + 1)}{384 \times 1200} = 243 = \text{F3H}$$

主程序:

```
MOV    TMOD, #20H    ;定时器 1 设为模式 2
MOV    TL1, #0F3H    ;定时器初值
MOV    TH1, #0F3H    ;8 位重装值
SETB   TR1           ;启动定时器 1
MOV    SCON, #50H    ;设置为方式 1, REN=1
MOV    R0, #20H      ;发送数据区首址
MOV    R1, #40H      ;接收数据取首址
ACALL  SOUT          ;先输出一个字符
SETB   ES
SETB   EA
SJMP   $             ;等待中断
```

中断服务子程序:

```
ORG    0023H        ;串行口中断入口
AJMP   SBR1          ;转至中断服务程序
ORG    0100H
```

```

SBR1: JNB    RI,SEND    ;TI=1,为发送中断
      ACALL  SIN        ;RI=1,为接收中断
      SJMP   NEXT       ;转至统一的出口
SEND: ACALL  SOUT       ;调用发送子程序
NEXT: RETI             ;中断返回

```

发送子程序:

```

SOUT: CLR   TI
      MOV   A,@R0      ;取数
      MOV   C,P
      CPL   C          ;奇校验
      MOV   ACC.7,C
      INC   R0         ;改指针
      MOV   SBUF,A     ;发送
      RET          ;返回

```

接收子程序

```

SIN:  CLR   RI
      MOV   A,SBUF     ;读
      MOV   C,P        ;取校验位
      CPL   C          ;奇校验
      ANL   A,#7FH     ;删校验位
      MOV   @R1,A;读
      INC   R1         ;改指针
      RET          ;返回

```

9.在 89C51 单片机系统中,采用查询方式由串行口发送带奇偶校验位的数据块,发送数据为内部 RAM 单元 20H-3FH 中的 ASCII 码数据,波特率为 1200b/s,晶振频率为 11.059MHz。

解: 本题由内部 RAM 单元 20H-3FH 取出 ASCII 码数据,在最高位上加奇偶校验位后由串行口发出。采用 8 位异步通信方式,波特率为 1200b/s, $f_{osc}=11.059\text{MHz}$ 。

由要求可知,应把串行口设置为方式 1,采用定时器 1 模式 2 作为波特率发生器,预置值 $(TL1) = (TH1) = 0E8H$ 。

$$X = 256 - \frac{11.059 \times 10^6 \times (0+1)}{32 \times 12 \times 1200} \cong 232 = 0E8H$$

主程序;

```

      MOV   TMOD,#20H    ;设置定时器 1 为模式 2
      MOV   TL1,#0E8H    ;初值,波特率为 1200b/s
      MOV   TH1,#0E8H
      SETB  TR1          ;启动 T1 运行
      MOV   SCON,#01000000B ;设置串行口为方式 1
      MOV   R0,#20H
      MOV   R7,#32       ;数据块长度
LOOP: MOV   A,@R0
      ACALL SP-OUT

```

```

JNB    P ,ERROR          ;出错
INC     R0
DJNZ    R7 ,LOOP

```

.....

串行口发送子程序（奇校验）；

SP-OUT:

```

MOV    C ,PSW.0          ;P
CPL    C                  ;奇校验
MOV    ACC.7 ,C
MOV    SBUF ,A            ;启动发送
CLR    TI                 ;清 TI
RET

```

ERROR: （略）

10.在 89C51 单片机系统中由串行口接收带奇偶校验位的数据块，采用查询方式，接收器把接收到的 32B 数据存放在 20H-2FH 单元内，波特率为 1200b/s，晶振频率为 11.059MHz，若奇偶校验出错则置进位位为 1。

解：

由要求可知，应把串行口设置为方式 1，采用定时器 1 模式 2 作为波特率发生器，预置值（TL1）=（TH1）=0E8H。

$$X = 256 - \frac{11.059 \times 10^6 \times (0 + 1)}{32 \times 12 \times 1200} \cong 232 = 0E8H$$

程序清单：

```

MOV    SCON ,#01010000B;方式 1，允许接收
MOV    TMOD ,#20H      ;设定定时器 T1 模式 2
MOV    TL1 ,#0E8H      ;初值，波特率 1200b/s
MOV    TH1 ,#0E8H
SETB   TR1              ;启动 T1 运行
MOV    R0 ,#20H
MOV    R7 ,#32          ;数据块长度
LOOP:  ACALL  SP-IN      ;调接收一帧子程序
JC     ERROR            ;由 SP-IN 中结果决定
MOV    @R0 ,A           ;存放接收的数据
INC     R0
DJNZ    R7 ,LOOP

```

.....

接收一帧子程序；

```

SP-IN: JNB   RI ,\$      ;RI 由硬件置位
CLR    RI               ;软件清除 RI
MOV    A ,SBUF
MOV    C ,P             ;检查奇校验位
CPL    C                ;置 C 为主程序用

```




```

        ANL  A,#7FH           ;去掉奇校验位
        RET
ERROR:   (略)

```

11.在 89C51 单片机系统中用第 9 个数据位作奇偶校验位，设计一个发送程序，将片内 RAM 50H~5FH 中的数据串行发送。编制串行口方式 2 的发送程序，波特率为 $f_{osc}/32$ ，晶振频率为 6MHz。

解：程序流程图



解：串行口设定为方式 2 状态，TB8 作奇偶校验位。在数据写入发送缓冲器之前，先将数据的奇偶位 P 写入 TB8，这时，第 9 位数据作奇偶校验用。

方式 2 发送程序流程图如图所示。

程序清单如下：

```

TRT:  MOV  SCON,#80H           ;方式 2 设定
      MOV  PCON,#80H           ;取波特率为 fosc/32
      MOV  R0,#50H             ;首址 50H 送 R0
      MOV  R7,#10H             ;数据长度 10H→R7
LOOP: MOV  A,@R0               ;取数据→A
      MOV  C,PSW.0             ;P→TB8
      MOV  TB8,C
      MOV  SBUF,A              ;数据→SBUF,启动发送
WAIT:  JBC  TI,CONT            ;判断发送中断标志

```

```

        SJMP WAIT
CONT: INC R0
        DJNZ R7,LOOP
        RET

```

12.在 89C51 单片机系统中,设计一个接收程序,接收到的 32B 数据存放在 20H-2FH 单元中。编制串行口方式 2 的接收程序,波特率为 $f_{osc}/32$,晶振频率为 6MHz,并核对奇偶校验位。
解:在方式 2、方式 3 的发送过程中,将数据和附加在 TB8 中的奇偶位一块发向对方。因此,作为接收的一方应设法取出该奇偶位进行核对,相应的接收程序段为:

```

RRR:    MOV  SCON,#90H    ;选方式 2,并允许接收 (REN=1)
LOOP:   JBC  RI,RECEIV    ;等待接收数据并清 RI
        SJMP LOOP
RECEIV: MOV  A,SBUF        ;将接收到的字符取出后,送到 ACC。
        ;注意,传送指令影响 PSW,产生接收端的奇偶值
        JB  PSW.0,ONE      ;判断接收端的奇偶值
        JB  RB8,ERR        ;判断发送端的奇偶值
        SJMP RIGHT
ONE:    JNB  RB8,ERR
RIGHT:  ;接收正确
        .....
ERR:    ..... ;接收有错

```

提示:当接收到一个字符时,从 SBUF 转移到 ACC 中时会产生接收端的奇偶值,而保存在 RB8 中的值为发送端的奇偶值,两个奇偶值应相等,否则接收字符有错。发现错误要及时通知对方重发。

13 在 89C51 单片机系统中,编制一个接收程序,将接收的 16 字节数据送入片内 RAM 的 50H-5FH 单元中。设串行口工作于方式 3,波特率为 2400b/s,晶振频率为 11.059MHz。
解:方式 3 为 11 位异步通信方式,波特率取决于 TI 的溢出率。查表 7-2 可知,当晶振为 11.059MHz,波特率为 2400b/s 时,可取 SMOD=0。

```

MAIN: MOV  TMOD,#20H      ;设 TI 工作于模式 2
        MOV  TH1,#0F4H    ;赋循环计数初值
        MOV  TL1,#0F4H    ;赋计数值
        SETB TR1          ;启动定时器 T1
        MOV  R0,#50H      ;首地址送 R0
        MOV  R7,#10H      ;数据长度送 R7
        MOV  SCON,#0D0H   ;串行口工作与方式 3,可接收
        MOV  PCON,#00H    ;设 SMOD=0
WAIT:   JBC  RI,PR1        ;接收完一帧数据,清 RI,转 PR1
        SJMP WAIT          ;否则等待
PR1:    MOV  A,SBUF        ;读入数据
        JNB  P,PNP        ;P=0,转 PNP
        JNB  RB8,PER       ;P=1,RB8=0,转出错处理
        SJMP RIGHT

```

```

PNP:   JB    RB8 ,PER           ;P=0, RB8=1, 转出错处理
RIGHT: MOV   @R0 ,A           ;数据送内存
        INC   R0               ;修改地址指针
        DJNZ  R7 ,WAIT        ;数据未接收完, 继续接收下一个数据
        CLR   PSW.5           ;置正确接收完毕标志 F0=0
        RET
PER:    SETB  PSW.5
        RET

```

14. 以 89C51 串行口按工作方式 1 进行串行数据通信。假定晶振频率为 6MHz，波特率为 1200b/s，以中断方式传送数据。请编写全双工通信程序。

解：设系统时钟频率 $f_{osc}=6.0\text{MHz}$ 。可取 SMOD=0, T1 的计数初值为 F3H。程序如下：

```

ORG 0000H
AJMP MAIN           ;上电, 转向主程序
ORG 0023H           ;串行口的中断入口地址
AJMP SERVE          ;转向中断服务程序
ORG 0040H           ;主程序
MAIN: MOV  SP,#60H   ;设置堆栈指针
      MOV  SCON ,#50H
      MOV  PCON ,#00H
      MOV  TMOD,#20H
      MOV  TH1,#0F3H
      MOV  TL1,#0F3H
      SETB TR1
      MOV  R0 ,#20H   ;置发送数据区首地址
      MOV  R1 ,#40H   ;置接收数据区首地址
      MOV  R7 ,#10H   ;置发送字节长度
      MOV  R6 ,#10H   ;置接收字节长度
      SETB ES         ;允许串行口中断
      SETB EA         ;CPU 允许中断
      MOV  A ,@R0     ;取第一个数据发送
      MOV  SBUF ,A    ;发送第一个数据
      SJMP $          ;等待中断
SERVE: JNB  RI ,SEND   ;TI=1,为发送中断
      CLR  RI
      MOV  A ,SBUF     ;读出接收缓冲区内容
      MOV  @R1 ,A      ;读入接收缓冲区
      DJNZ R6 ,L1      ;判断数据块发送完否
      SJMP L2          ;数据块接收完, 转 L2
L1:   INC  R1          ;修改数据区指针
L2:   RETI            ;中断返回
SEND:
      CLR  TI          ;清除发送中断标志
      DJNZ R7 ,L3      ;判断数据块发送完否

```

```

        SJMP L4                ;数据块接收完，转 L4
L3:    MOV  A,@R0              ;取数据发送
        MOV  SBUF,A           ;发送数据
        INC  R0               ;修改数据地址
L4:
        RETI                  ;中断返回
        END

```

15. 以 89C51 单片机串行口按工作方式 3 进行串行数据通信。假定波特率为 1200b/s，晶振频率为 11.059MHz，第 9 数据位作奇偶校验位，以中断方式传送数据。请编写通信程序。

解：

```

        ORG  0000H
        AJMP MAIN              ;上电，转向主程序
        ORG  0023H              ;串行口的中断入口地址
        AJMP STOP              ;转向中断服务程序
        ORG  0040H              ;主程序
MAIN:   MOV  SP,#60H
        MOV  TMOD,#20H
        MOV  TH1,#0E8H
        MOV  TL1,#0E8H
        SETB TR1
        MOV  SCON,#0D0H
        MOV  PCON,#00H
        MOV  R0,#20H           ;置发送数据区首地址
        MOV  R1,#40H           ;置接收数据区首地址
        SETB ES                ;允许串行口中断
        SETB EA                ;CPU 允许中断
        MOV  A,@R0
        MOV  C,PSW.0           ;P→C
        CPL  C
        MOV  TB8,C
        MOV  SBUF,A            ;发送第一个数据
        SJMP $
STOP:   JNB  RI,SOUT            ;TI=1,为发送中断
        CLR  RI
        MOV  A,SBUF            ;读出接收缓冲区内容
        MOV  C,PSW.0           ;P→C
        CPL  C                 ;形成奇校验
        JC   LOOP1             ;判断接收端的奇偶值，C=1转 LOOP1
        JNB  RB8,LOOP2         ;C=0, RB8=0, 转 LOOP2
        SJMP ERROR             ;C=0,RB8=1,转出错处理
LOOP1:  JB   RB8,LOOP2         ;C=1, RB8=1, 转 LOOP2
        SJMP ERROR             ;C=0,RB8=1,转出错处理
LOOP2:

```

```

        MOV @R1 ,A      ;将接收数据送入接收数据区
        INC  R1          ;修改数据区指针
        RETI
SOUT: CLR  TI           ;是发送中断，清除发送中断标志
        INC  R0          ;修改数据区指针
        MOV  A,@R0
        MOV  PSW.0, C    ;P→C
        CPL  C
        MOV  TB8 ,C
        MOV  SBUF ,A     ;发送第一个数据
        RETI
ERROR:.....

```

期中测验试卷

单选题（共 20 题，每小题 5 分，共 100 分）

- 1、二进制数 100111100B 转换成十六进制数（ D ）。
A. 12C H B. 12E H C. 13A H D. 13C H
- 2、将十进制数 0.625 转换成二进制数（ B ）。
A. 0.011B B. 0.101B C. 0.1001B D. 0.1011B
- 3、将十六进制数 2AFH 用二进制数表示为（ B ）。
A. 001010100111B B. 001010101111B C. 001010101011B D. 001010101101B
- 4、十进制数 253 化成十六进制数为（ D ）。
A. 0EBH B. 0EDH C. 0FBH D. 0FDH
- 5、在 89C51 单片机中，表示负数通常使用（ D ）。
A. 无符号数 B. 原码 C. 反码 D. 补码
- 6、89C51 单片机的一个机器周期有（ B ）个时钟周期，若 $f_{osc}=12\text{MHz}$ ，则一个机器周期为（ ） μs 。
A. 6, 1 B. 12, 1 C. 6, 2 D. 12, 2
- 7、89C51 单片机中复位后正确的是（ C ）。
A. 内部寄存器内容全为 00H
B. SP 的内容为 60H
C. P0~P3 的内容为 0FFH
D. SBUF 内容为 0FFH
- 8、89C51 单片机中在堆栈中弹出一个数据时（ A ）。
A. 先出栈，再令 SP-1 B. 先令 SP-1，再出栈
C. 先出栈，再令 SP+1 D. 先令 SP+1，再出栈
- 9、89C51 单片机中下列正确指令中源操作数属于直接寻址的是（ D ）。
A. MOV 30H,A B. MOVX A,@R0C. MOV P0,#30HD. MOV A,P0
- 10、89C51 单片机中设（A）=53H，（R0）=0FCH，执行指令 ADD A, R0；结果（ B ）。
A. （A）=4FH, CY=0, AC=0, OV=0, P=1
B. （A）=4FH, CY=1, AC=0, OV=0, P=1
C. （A）=34H, CY=1, AC=0, OV=0, P=1
D. （A）=4FH, CY=1, AC=0, OV=1, P=1
- 11、89C51 单片机中，若：（R1）=30H （30H）=11H，求执行下面指令后的结果（ D ）。
INC @R1;
INC R1 ;
A. （30H）=11H, （R1）=30H
B. （30H）=12H, （R1）=30H
C. （30H）=11H, （R1）=31H
D. （30H）=12H, （R1）=31H
- 12、89C51 单片机中假定累加器 A 中的内容为 30H，执行指令
1000H: MOVC A,@A+PC
后，把程序存储器（ C ）单元的内容送入累加器 A 中。
A. 1000H B. 1030H C. 1031H D. 1032H
- 13、89C51 单片机中执行下列 3 条指令后，30H 单元的内容是（ C ）。
MOV R0, #30H
MOV 40H, #0EH

MOV @R0, 40H

A. 40H B. 30H C. 0EH D. FFH

14、89C51 单片机中执行 7H 和 5BH 两个立即数相加后，和为 (C)，(CY) 是 ()。

A. 1CH, 0 B. C1H, 0 C. 02H, 1 D. C1H, 1

15、89C51 单片机中若 (A)=86H, (PSW)=80H, 则执行 RRC A 指令后，(A) 为 (A)。

A. C3H B. B3H C. 0DH D. 56H

16、89C51 单片机中如果 (P0)=65H，则当

CPL P0.2

SETB C

MOV P0.4, C

执行后 (P0) = (C)

A. 61H B. 75H C. 71H D. 17H

17、89C51 单片机中，若初始 ROM 内容为：

2000H	00H
2001H	01H
2002H	09H
2003H	04H
2004H	06H
2005H	08H
2006H	FFH
2007H	66H
2008H	45H
2009H	ABH
200AH	11H

执行下面程序段后 A 中内容是什么？

MOV DPTR, #2000H ; (DPTR)=2000H

MOV A, #0AH ; (A)=0AH

MOVC A, @A+DPTR ;

(A) = (A)

A. 11H B. 0AH C. ABH D. FFH

18、89C51 单片机中计算下面子程序中指令的偏移量和程序执行的时间(晶振频率为 6MHz)。

(C)

DEL: MOV R7, #200

DEL1: MOV R6, #125

DEL2: DJNZ R6, DEL2

DJNZ R7, DEL1

RET

A. 延时约 200ms B. 延时约 125ms C. 延时约 100ms D. 延时约 50ms

19、89C51 单片机中，编写程序进行数据块传送。将片内 RAM 20H 单元开始的 10 个字节，传送到片外 RAM 1000H 单元开始的单元中去。请补全括号中代码。

ORG 0800H

START: MOV R0, #20H

MOV DPTR, #1000H

MOV R7, #0AH

```
LOOP:  MOV  A,( B )
        MOVX  @DPTR,A
        INC  R0
        INC  DPTR
        DJNZ  ( ),LOOP
        SJMP  $
```

A. R0,R7 B. @R0,R7 C. R0,@R7 D. @R0,@R7

20、89C51 单片机中，编写程序将(30H)，(31H)单元中的数与(40H)，(41H)单元中的数相加，结果存于(30H)，(31H)单元中。请补全括号中代码。

```
ORG 2000H
```

```
MOV  A, 30H
```

```
ADD  A, 40H
```

```
MOV  30H, A
```

```
MOV  A, ( D )
```

```
ADDC A, ( )
```

```
MOV  31H, A
```

A. 30H, 40H B. 30H, 41H C. 31H, 40H D. 31H, 41H

期末模拟试卷

一、单项选择题(每小题 2 分，共 20 分，答案填入下面答题区)

1	2	3	4	5	6	7	8	9	10

1. 10101.101B 转换成十进制数是 (D)。

- A. 46.625 B. 23.625 C. 23.62 D. 21.625

2. 73.5 转换成十六进制数是 (B)。

- A. 94.8H B. 49.8H C. 111H D. 49H

3. 3D.0AH 转换成二进制数是 (A)。

- A. 111101.0000101B B. 111100.0000101B
C. 111101.101B D. 111100.101B

4. 十六进制数 1C5H 转化成十进制数为 (D)。

- A. 447D B. 449D C. 451D D. 453D

5. 89C51 单片机的一个机器周期有 () 个振荡周期，若 $f_{osc}=12\text{MHz}$ ，则一个机器周期为 (B) μs 。

- A. 6, 1 B. 12, 1 C. 6, 2 D. 12, 2

6. 89C51 单片机中，若：(A)=78H，(R0)=64H，执行”ADD A, R0”后，PSW= (B)。

- A. 04H B. 05H C. 08H D. 0AH

7. 如果 (P0)=65H，则当

CPL P0.2

SETB C

MOV P0.4, C

执行后 (P0) = (C)

- A. 61H B. 75H C. 71H D. 17H

8. 89C51 单片机中设串行口工作于方式 1，晶振频率为 6MHz，波特率为 1200 位/秒，SMOD=0，定时器 1 工作于方式 2，则定时器 1 的计数初值为 (C)。

- A. F0H B. F1H C. F3H D. F4H

9. 89C51 单片机的时钟频率为 6MHz，若要求定时 1ms，定时/计数器工作于模式 1，其定时/计数器的初值为 (A)

- A. FE0CH B. FE0BH C. FE0AH D. FE09H

10. 单片机 89C51 的时钟频率为 12MHz，计算下面子程序中指令的偏移量和程序执行的时间。(D)

```

MOV    R3,#15      ;1 个机器周期
DL1:MOV R4,#255     ;1 个机器周期
DL2:MOV P1,R3       ;2 个机器周期
      DJNZ  R4,DL2   ;2 个机器周期
      DJNZ  R3,DL1   ;2 个机器周期
      RET           ;2 个机器周期

```

A.15345us

B. 15346us

C.15347us

D. 15348us

分析： $((2+2) \times 255 + 1 + 2) \times 15 + 1 + 2 = 15348\text{us}$

二、算法程序编程(每小题 10 分，共 50 分，本题应用汇编语言编程)

1. 在 89C51 单片机中进行数据块传送。将片内 RAM 40H 单元开始的 50 个字节，传送到片外 RAM 2000H 单元开始的单元中去。

解：

```

ORG    0800H
START: MOV    R0, #40H
      MOV     DPTR, #2000H
      MOV     R7, #32H
LOOP:  MOV     A, @R0
      MOVX    @DPTR, A
      INC     R0
      INC     DPTR
      DJNZ    R7, LOOP
      SJMP    $

```

2. 在 89C51 单片机中将片内 RAM 30H 单元的内容拆成两段，每段四位。并将它们分别存入 31H 与 32H 单元中。

解：

```

ORG    2000H
START: MOV    R0, #31H
      MOV     A, 30H
      ANL     A, #0FH
      MOV     @R0, A
      INC     R0
      MOV     A, 30H
      SWAP    A
      ANL     A, #0FH
      MOV     @R0, A
      SJMP    $

```

3. 在 89C51 单片机中设被加数存放在内部 RAM 的 40H、41H 单元，加数存放在 42H、43H 单元，若要求和存放在 44H、45H 中，试编写出 16 位无符号数相加的程序。

答：程序如下：

```

ORG    0000H
MOV     R0, #41H

```

```

MOV    R1,#43H
MOV    A,@R0
ADD    A,@R1
MOV    45H,A
DEC    R0
DEC    R1
MOV    A,@R0
ADDC   A,@R1
MOV    44H,A
SJMP   $
END

```

4.在 89C51 单片机中编写延时 6 ms 的子程序。设晶振频率为 12MHz。

答：晶振频率为 12MHz，则 1 个机器周期为 1 微妙（1μs），

一条 DJNZ 指令为 2 个机器周期，则执行该指令 3000 次为 6ms，3000=30*100

```

ORG    1000H
Delay:  MOV    R6,#30
DLY1:   MOV    R7,#100
DLY2:   DJNZ   R7,DLY2
DJNZ   R6,DLY1
RET

```

5. 在 89C51 单片机中编写程序实现如下功能：设变量 x 存放于 VAR 单元（60H）中，函数值 y 存放于 FUNC（61H）中，按下式赋值：

$$y = \begin{cases} 1 & x > 1 \\ 0 & x = 1 \\ -1 & x < 1 \end{cases}$$

解：程序如下：

```

ORG    0800H
VAR    DATA    60H           ;定义 VAR
FUNC   DATA    61H           ;定义 FUNC
START: MOV    A, VAR           ;取 x
      CJNE   A,#01H,COMP1      ;不为 1 转 COMP1
      MOV    A,#00H
      SJMP   COMP
COMP1: JNB    ACC.7, POSI       ;x>0 转 POSI
      MOV    A,#0FFH           ;x<0,-1→A
      SJMP   COMP
POSI:  MOV    A,#01H
COMP:  MOV    FUNC,A
      END

```

三、应用程序编程 （每小题 10 分，共 30 分，本题应用汇编语言或 C 编程。）

1. 在 89C51 单片机中以中断方法使用定时器 T0 设计程序实现如下功能;假定 P1.0 每 10ms 产生一个机器周期的正脉冲（正脉冲宽度为 1 个机器周期），P1.1 每 600ms 产生一个机器周期的正脉冲（正脉冲宽度为 1 个机器周期）。设晶振频率为 12MHz，其它参数可自行选择，需给出参数设计说明。

解: (1) 实现时钟计时的基本方法

1) 计算计数初值。

把定时器的定时时间定为 10ms，计数溢出 60 次即得到 600ms；60 次计数可用软件方法实现。

假定使用定时器 T0，以工作模式 1 进行 10ms 的定时。如 $f_{osc}=12\text{MHz}$ ，则计数初值 X 为：

$X=55536=D8F0H$ ，因此：(TL0)=F0H (TH0)=D8H


2) 采用定时方式进行溢出次数的累计，计满 60 次即得到 600ms 计时。

设置软件计数器初值为 60，每 10ms 定时时间到溢出中断，使软件计数器减 1，直到减到 0，则 600ms 到。

(2) 到了这些时刻，分别在 P1.0、P1.1 输出正脉冲即可。

```
ORG 0000H
AJMP START
ORG 000BH      ;定时器 T0 中断入口.
AJMP SERVE
START: MOV TMOD, #01H    ;T0 定时方式 1
      MOV TH0, #0D8H     ;定时
      MOV TL0, #0F0H
      SETB TR0           ;启动.
      SETB ET0           ;允许中断.
      SETB EA
      MOV R3, #60        ;600ms.
      CLR P1.0           ;开始时，输出低电平.
      CLR P1.1
      SJMP $             ;等待中断.
SERVE: ;10ms 执行一次本中断程序.
      MOV TL0, #0F0H     ;重新送入初始值.
      MOV TH0, #0D8H     ;定时
      SETB P1.0          ;到了 10ms，输出正脉冲.
      CLR P1.0           ;马上就恢复为 0，这就是 1 个机器周期.
      DJNZ R3, T0_END    ;不到 600ms 转移.
      SETB P1.1          ;到了 600ms，输出正脉冲.
      CLR P1.1           ;马上就恢复为 0，这就是 1 个机器周期.
      MOV R3, #60
T0_END: RETI
      END
```

2. 89C51 单片机应用系统的晶荡频率为 12MHz，请利用定时器 T1 编程实现在 P1.7 引脚输出频率为 20Hz 的方波。要求：采用中断方式，T1 工作在方式 1 下。需给出参数设计说明。



解:

1) 确定定时器 T1 初值

∴ 方波周期 $T=1/20=0.05s=50ms$

∴ 用 T1 定时 50ms, 时间到 P1.7 引脚电平取反

定时时间: $t = (2^{16} - T1\text{初值}) \times \text{振荡周期} \times 12$

$$T1\text{初值} = 2^{16} - \frac{t}{\text{振荡周期} \times 12}$$

$$\therefore T1\text{初值} = 2^{16} - \frac{25ms}{\frac{1}{12 \times 10^6} \times 12}$$

$$\therefore T1\text{初值} = 40536 = 9E58H$$

$$\therefore (TH1) = 9EH, (TL1) = 58H$$

2) 确定工作模式寄存器 TMOD 的值

∴ 定时器 T1 工作于模式 1 的定时器工作方式,

∴ 高四位: GATE=0, C/T=0, M1M0=01, 低四位: 取 0。

∴ (TMOD) = 0001 0000 B = 10H


3) 编程:

```

ORG    0000H
LJMP   START
ORG    001BH
LJMP   TS
ORG    0080H
START: MOV    TMOD,#10H           ;T1 为模式 1
      MOV     TL1,#9EH           ;送初值
      MOV     TH1,#58H
      SETB    P1.7              ;P1.7 置 1
      SETB    TR1               ;启动定时
      SETB    ET1
      SETB    EA
      SJMP    $
TS:    CLR     TF1               ;产生中断, 清标志位
      MOV     TL1,#9EH           ;重新置值
      MOV     TH1,#58H
      CPL     P1.7              ;取反
      RETI                     ;重复循环
      END

```

3.在 89C51 单片机系统中用第 9 个数据位作奇偶校验位，设计一个接收程序，将片内 RAM 50H~6FH 中的数据串行发送。编制串行口方式 2 的接收程序，波特率为 $f_{osc}/64$ ，晶振频率为 12MHz。要求画出程序流程图。



程序如下：

```

ORG 0000H
LJMP TRT
TRT: MOV SCON, #90H      ;方式2设定，REN=1
      MOV PCON, #00H     ;取波特率为 fosc/64
      MOV R0, #50H       ;首地址→R0
      MOV R7, #20H       ;数据长度 20H→R7
WAIT: JBC RI, CONT       ;判断接收中断标志
      SJMP WAIT
CONT: CLR RI
      MOV A, SBUF         ;读出接收缓冲区内容
      MOV @R0, A
      MOV C, RB8          ;RB8→C
      MOV 20H, C          ;C→20H
      INC R0
      DJNZ R7, WAIT
  
```