



武汉纺织大学
WUHAN TEXTILE UNIVERSITY

单片机原理及应用

课程性质：必修（考试） 学时：48 学分：3
考核方式：闭卷考试



第六章 定时器及应用

§ 6.1 定时器概述

§ 6.2 定时器的控制

§ 6.3 定时器的四种模式及应用

§ 6.4 思考题与习题

§ 6.1 定时器概述



§ 6.1.1 什么是计数和定时

§ 6.1.2 89C51定时器组成



§ 6.1.1 什么是计数和定时


什么是计数？计数是指对外部事件进行计数。

什么是定时？定时是通过计数器来实现的，不过此时的计数脉冲来自单片机内部，即每个机器周期产生一个计数脉冲，也就是每个机器周期计数器加1



§ 6. 1. 2 89C51定时器组成

- 89C51定时器的结构如图6-2所示。
- 有两个16位的定时器/计数器，即定时器0（T0）和定时器1（T1）。都是16位加1计数器。
- T0由两个8位特殊功能寄存器TH0和TL0构成；T1由TH1和TL1构成。



§ 6. 1. 2 89C51定时器组成



- 每个定时器都可由软件设置为**定时工作方式**或 **计数工作方式**。
由特殊功能寄存器TMOD和TCON所控制。
- 定时器工作不占用CPU时间，除非定时器/计数器溢出，才能中断CPU的当前操作。
- 定时器/计数器有四种工作模式。其中模式0-2对T0和T1是一样的，模式3对两者不同。



定时工作方式

定时器计数89C51片内振荡器输出经12分频后的脉冲，即每个机器周期使定时器（T0或T1）的数值加1直至计满溢出。

当89C51采用12MHz晶振时，一个机器周期为 $1\mu\text{s}$ ，计数频率为1MHz。



计数工作方式

- 通过引脚T0（P3.4）和T1（P3.5）对外部脉冲信号计数。当输入脉冲信号产生由1至0的下降沿时计数器的值加1。
- CPU检测一个1至0的跳变需要两个机器周期，故最高计数频率为振荡频率的1/24。
- 为了确保某个电平在变化之前被采样一次，要求电平保持时间至少是一个完整的机器周期。
- 对输入脉冲信号的基本要求如图6-2所示。



§ 6.2 定时器的控制

§ 6.2.1 工作模式寄存器 TMOD

§ 6.2.2 控制器寄存器 TCON




§ 6.2.1 工作模式寄存器 TMOD

- TMOD用于控制T0和T1的工作模式。
- TMOD不能位寻址，只能用字节设置定时器的的工作模式，低半字节设置T0，高半字节设置T1。
- 89C51系统复位时，TMOD的所有位被清0。
- TMOD各位的定义格式如 图6-4 所示。
- TMOD各位定义及具体的意义归纳如图6-5所示。



图6-4 工作模式寄存器TMOD的位定义





§ 6.2.1 工作模式寄存器 TMOD

① M1和M0—操作模式控制位。

两位可形成四种编码，对应于四种模式。

表6-1 M1, M0控制的四种工作模式

M1	M0	工作模式	功能描述
0	0	模式 0	13 位计数器
0	1	模式 1	16 位计数器
1	0	模式 2	自动再装入 8 位计数器
1	1	模式 3	定时器 0：分成二个 8 位计数器 定时器 1：停止计数



§ 6.2.1 工作模式寄存器 TMOD

② C/\overline{T} —计数器/定时器方式选择位。

$C/\overline{T}=0$ ，设置为**定时方式**。定时器计数89C51片内脉冲，即对机器周期计数。

$C/\overline{T}=1$ ，设置为**计数方式**。计数器的输入来自引脚T0（P3.4）或T1（P3.5）端的外部脉冲。



§ 6.2.1 工作模式寄存器 TMOD

③ GATE—门控位。

GATE=0，只要用软件使TR0（或TR1）置1就可以启动定时器，而不管 $\overline{\text{INT0}}$ （或 $\overline{\text{INT1}}$ ）的电平是高还是低。

GATE=1，只有 $\overline{\text{INT0}}$ （或 $\overline{\text{INT1}}$ ）引脚为高电平且由软件使TR0（或TR1）置1时，才能启动定时器工作。



§ 6.2.2 控制器寄存器 TCON

- TCON除可字节寻址外，各位还可位寻址。
- 89C51系统复位时，TCON的所有位被清0。
- TCON各位的定义格式如 图6-7所示。
- TCON各位定义及具体的意义归纳如图6-8所示。




§ 6.2.2 控制器寄存器 TCON

	8FH	8EH	8DH	8CH	8BH	8AH	89H	88H
TCON (88H)	TF1	TR1	TF0	TR0	IE1	IT1	IE0	IT0

图6-7 控制寄存器TCON的位定义



图6-8 TCON各位定义及具体的意义





① TF1 (TCON. 7) —T1溢出标志位。

当T1溢出时，由硬件自动使中断触发器TF1置1，并向CPU申请中断。

当CPU响应中断进入中断服务程序后，TF1被硬件自动清0。
TF1也可以用软件清0。



② TF0 (TCON. 5) —T0溢出标志位。

其功能和操作情况同TF1。

当**T0溢出**时，由硬件自动使中断触发器**TF0置1**，并向CPU申请中断。

当**CPU响应中断进入中断服务程序后**，TF1被硬件**自动清0**。
TF1也可以用**软件清0**。



③ TR1 (TCON. 6)—T1运行控制位。

可通过软件置1（TR1=1）或清0（TR1=0）来启动或关闭 T1工作。

在程序中用指令“SETB TR1”使TR1位置1，定时器T1便开始计数。用“CLR TR1”使TR1清0，定时器停止工作。



④ TR0 (TCON. 4)—T0运行控制位。

其功能和操作情况同TR1。

可通过软件置1（TR0=1）或清0（TR0=0）来启动或 关闭 T0。

在程序中用指令“**SETB TR0**”使TR0位置1，定时器T0便开始计数。用“**CLR TR0**”使TR0清0，定时器停止工作。



⑤ IE1, IT1, IE0, IT0 (TCON. 3~TCON. 0)

——外部中断INT1, $\overline{\text{INT0}}$ 请求及请求方式控制位。
前一章已经讲过。



§ 6.3 定时器的四种模式及应用

§ 6.3.0 模式 0 及其应用

§ 6.3.1 模式 1 及其应用

§ 6.3.2 模式 2 及其应用

§ 6.3.3 模式 3 及其应用

§ 6.3.4 综合应用举例



§ 6.3.0 模式 0 及其应用

一、模式 0 的逻辑电路结构

T0在模式 0 的逻辑电路结构如图6-8所示。（T1相同）


二、模式 0 工作特点

三、模式 0 的应用举例



二、模式 0 工作特点

- 在这种模式下，16位寄存器（TH0和TL0）只用了13位。其中TL0的高3位未用，其余5位为整个13位的低5位，TH0占高8位。当TL0的低5位溢出时，向TH0进位；TH0溢出时，向中断标志TF0进位（硬件置位TF0），并申请中断。



- 定时时间为： $t = (2^{13} - T0\text{初值}) \times \text{振荡周期} \times 12$
- 用于计数工作方式时，计数长度为： $L = (2^{13} - T0\text{初值})$ （个外部脉冲）



三、模式 0 的应用举例

【例】：设晶振为12MHz，试计算定时器T0工作于模式0时的最大定时时间T。

解：当T0处于工作模式0时，加1计数器为13位。

因为：定时时间为： $t = (2^{13} - \text{T0初值}) \times \text{振荡周期} \times 12$

最大定时时间为“T0初值=0”时。

所以：

$$\begin{aligned} T &= 2^{13} \times \text{振荡周期} \times 12 \\ &= 2^{13} \times \frac{1}{12 \times 10^6} \times 12 \\ &= 8192 \times 10^{-6} \\ &= 8.192 \text{ ms} \end{aligned}$$



【例】： 设定定时器T0用于定时10ms，晶振为6MHz。试确定T0初值。并编写定时器T0初始化程序段。

解：1) 确定T0初值

当T0处于工作模式0时，加1计数器为13位。

定时时间：
$$T = (2^{13} - T0\text{初值}) \times \text{振荡周期} \times 12$$

所以：
$$10\text{ms} = (2^{13} - T0\text{初值}) \times \text{振荡周期} \times 12$$

$$T0\text{初值} = 2^{13} - \frac{10 \times 10^{-3} \times 6 \times 10^6}{12}$$

$$\therefore T0\text{初值} = 3192 = \text{C78H}$$

$$= \text{0110001111000 B}$$


即： (TH0) = 0110 0011 B = 63H (高8位)

(TL0) = 1 1000 B = 18H (低5位)

解:



2) 确定TMOD的初值



TMOD的低四位为T0的控制位

模式0: **M1M0=00**,

定时器方式: **C/T=0**,

门控位不受INT0的影响: **GATE=0**,

其余位: 为 **0**。

∴ 模式字为 **TMOD=0000 0000 B=00H**

解:



3) 编写定时器T0的初始化程序段

主程序:

```
                ORG      0100H
MAIN:          MOV      SP, #60H      ; 设堆栈指针
                MOV      TMOD, #00H   ; 选择工作模式
                MOV      TH0, #63H    ; 送初值
                MOV      TL0, #18H
                SETB     TR0          ; 启动定时
                ...
```

【例】：设定定时器T1工作于模式0，晶振为6MHz。编程实现：每10ms时间P1.0取反的程序。



解：1) 确定T1初值

T1处于工作模式0时，定时时间位10ms（同例2）。

定时时间： $T = (2^{13} - T1\text{初值}) \times \text{振荡周期} \times 12$

(TH1) = 0110 0011 B = 63H（高8位）

(TL1) = 1 1000 B = 18H（低5位）

2) 确定TMOD的初值

TMOD的高四位为T1的控制位

模式0：M1M0=00，定时器方式：C/T=0，

门控位不受INT0的影响：GATE=0，其余位：为 0。

∴ 模式字为 TMOD=0000 0000 B=00H



解：3) 编写程序

初始化引导程序：

```
ORG    0000H
RESET: AJMP  MAIN    ; 跳过中断服务程序区
ORG    001BH    ; 中断服务程序入口
AJMP   ITOP
```

主程序：

```
ORG    0100H
MAIN:  MOV   SP, #60H    ; 设堆栈指针
      MOV   TMOD, #00H   ; 选择工作模式
      MOV   TH1, #63H    ; 送初值
      MOV   TL1, #18H
      SETB  P1.0         ; P1.0置1
      SETB  TR1          ; 启动定时
      SETB  ET1          ; T1开中断
      SETB  EA           ; CPU开中断
      HERE: AJMP  HERE    ; 等待时间到，转
                        ; 入中断服务程序
```

中断服务程序

```
ORG    0150H
ITOP:  MOV   TL1, #18H    ; 重新装入初值
      MOV   TH1, #63H
      CPL   P1.0         ; P1.0取反
      RETI                ; 中断返回
```




【例】：利用T0的工作模式0产生定时，在P1.0引脚输出周期为10ms的方波。设晶振频率 $f_{osc}=12\text{MHz}$ 。编程实现其功能（分别采用查询方式和中断方式）。

解：

要在P1.0引脚输出周期为10ms的方波，只要使P1.0每隔5ms取反一次即可。

(1) 选择工作模式

T0的模式字为 $\text{TMOD}=00\text{H}$,

即： $\text{M1M0}=00$, $\text{C/T}=0$, $\text{GATE}=0$, 其余位为0。

(2) 计算5ms定时T0的初值

$$T = (2^{13} - \text{T0初值}) \times (1/12) \times 10^{-6} \times 12 = 5 \times 10^{-3} \text{ s}$$

$$\text{T0初值} = 3192 \text{ D} = \text{C78 H} = 01100011 \ 11000 \text{ B}$$

T0的低5位：11000 B=18H 即 $(\text{TL0})=18\text{H}$

T0的高8位：01100011 B=63H 即 $(\text{TH0})=63\text{H}$



(3) 采用查询方式的程序

程序清单:

```
ORG    0000H
LJMP   MAIN          ; 初始化引导程序
ORG    0080H
MAIN:  MOV    TMOD,#00H  ; 设置T0为模式0
      MOV    TL0,#18H   ; 送初值
      MOV    TH0,#63H
      SETB   P1.0
      SETB   TR0        ; 启动定时
LP:    JBC    TF0, NEXT  ; 查询定时时间到否
      SJMP   LP
NEXT:  MOV    TL0,#18H   ; 重装计数初值
      MOV    TH0,#63H
      CPL    P1.0       ; 取反
      SJMP   LP        ; 重复循环
```



(4) 采用定时器溢出中断方式的程序

```
ORG 0000H      ; 主程序
RESET: AJMP MAIN ; 跳过中断服务程序区
ORG 000BH      ; 定时器T0中断矢量
AJMP CTC0      ; 转入中断服务程序
ORG 0030H
MAIN: MOV TMOD,#00H; 设置T0为模式0
MOV TL0,#18H   ; 送初值
MOV TH0,#63H
SETB EA        ; CPU开中断
SETB ET0       ; T0中断允许
SETB TR0       ; 启动定时
HERE: SJMP HERE ; 等待中断
ORG 0120H      ; 中断服务程序
CTC0: MOV TL0,#18H ; 重新装入初值
MOV TH0,#63H
CPL P1.0       ; P1.0取反
RETI           ; 中断返回
```

模式 0 的应用举例完



§ 6.3.1 模式 1 及其应用

- 一、模式 1 的逻辑电路结构： T0在模式 1 的逻辑电路结构如图6-8所示。（T1相同）
- 二、模式 1 工作特点
- 三、模式 1 的应用举例



一、模式 1 的逻辑电路结构





图6-8 T0(T1)模式1结构-16计数器



二、模式 1 工作特点

- 该模式对应的是一个16位的定时器/计数器。



- 用于定时工作方式时，定时时间为：
$$t = (2^{16} - T0\text{初值}) \times \text{振荡周期} \times 12$$
- 用于计数工作方式时，计数长度为：
$$(2^{16} - T0\text{初值}) \text{ (个外部脉冲)}$$



三、模式 1 的应用举例

【例】：设定定时器T0选择工作模式1的计数器工作方式，其计数器初值为FFFFH，问此时定时器T0的实际用途是什么？

解：因其初值为FFFFH，只要随机外来一脉冲即可溢出，向CPU申请中断，故这一内部中断源实质上已作为外部中断源使用。因此此定时器T0并不用于定时或计数。



【例】：设晶振为12MHz，试计算定时器T0工作于模式1时的最大定时时间T。

解：当T0处于工作模式1时，加1计数器为16位。

定时时间为： $t = (2^{16} - \text{T0初值}) \times \text{振荡周期} \times 12$

最大定时时间为“T0初值=0”时。

所以：

$$\begin{aligned} T &= 2^{16} \times \text{振荡周期} \times 12 \\ &= 2^{16} \times \frac{1}{12 \times 10^6} \times 12 \\ &= 65536 \times 10^{-6} \\ &= 65.536 \text{ ms} \end{aligned}$$




例6-1：用定时器T1产生一个50Hz的方波，由 P1.1输出。使用程序查询方式，
 $f_{osc}=12\text{MHz}$ 。

解：1) 确定定时器T1初值

∴ 方波周期 $T = 1/50 = 0.02\text{s} = 20\text{ms}$

∴ 用T1定时10ms, 时间到P1.1引脚电平取反。

定时时间： $t = (2^{16} - T1\text{初值}) \times \text{振荡周期} \times 12$



$$T1\text{初值} = 2^{16} - \frac{t}{\text{振荡周期} \times 12}$$

$$\therefore T1\text{初值} = 2^{16} - \frac{10\text{ms}}{\frac{1}{12 \times 10^6} \times 12}$$

$$\therefore T1\text{初值} = 55536 = \text{D8F0H}$$

$$\therefore (\text{TH1}) = \text{D8H}, (\text{TL1}) = \text{F0H}$$



解:

2) 确定工作模式寄存器TMOD的值

∴ 定时器T1工作于模式1的定时器工作方式,

∴ 高四位: GATE=0, C/T=0, M1M0=01, 低四位: 取0。

∴ (TMOD) = 0001 0000 B = 10H

3) 采用定时器查询方式的程序

```
                ORG  0000H
                LJMP  START

                ORG  0080H
START:  MOV  TMOD,#10H    ; T1为模式1
        MOV  TL1,#F0H    ; 送初值
        MOV  TH1,#0D8H
        SETB P1.1        ; P1.1置1
        SETB TR1         ; 启动定时
LOOP:   JNB  TF1, LOOP    ; 查询定时时间到否
        CLR  TF1         ;产生溢出, 清标志位
        MOV  TL1,#F0H    ; 重新置初值
        MOV  TH1,#0D8H
        CPL  P1.1        ; 取反
        SJMP LOOP        ; 重复循环
```



(4) 采用定时器溢出中断方式的程序

```
ORG 0000H      ; 主程序
RESET: AJMP MAIN ; 跳过中断服务程序区
ORG 001BH      ; 定时器T1中断矢量
AJMP IT1P      ; 转入中断服务程序
ORG 0030H
MAIN: MOV SP, #60H ; 设堆栈指针
      MOV TMOD, #10H ; 设置T1为模式1
      • MOV TL1, #F0H ; 置初值
      • MOV TH1, #0D8H
      SETB EA      ; CPU开中断
      SETB ET1     ; T1中断允许
      SETB TR1     ; 启动定时
      HERE: SJMP HERE ; 等待中断
      ORG 0120H    ; 中断服务程序
IT1P: MOV TL1, #F0H ; 重新装入初值
      MOV TH1, #D8H
      CPL P1.1     ; P1.1取反
      RETI        ; 中断返回
```

模式1的应用举例完



§ 6.3.2 模式 2 及其应用

- 一、模式 2 的逻辑电路结构
- 二、模式 2 工作特点
- 三、模式 2 的应用举例



一、模式 2 的逻辑电路结构




图6-9 T0(T1)模式2结构-16计数器



一、模式 2 的逻辑电路结构

- T0在模式 2 的逻辑电路结构如图6-9所示。（T1相同）
- TL0计数溢出时，不仅使溢出中断标志位TF0置1，而且还自动把TH0中的内容重新装载到TL0中。
- TL0用作8位计数器，TH0用以保存初值。



二、模式 2 工作特点

- 该模式把TL0 (TL1) 配置成一个可以自动重载的8位定时器/计数器。
- 在程序初始化时，TL0和TH0由软件赋予相同的初值。
- 用于定时工作方式时，定时时间为：
$$t = (2^8 - \text{TL0初值}) \times \text{振荡周期} \times 12$$
- 用于计数工作方式时，计数长度为：
$$(2^8 - \text{TL0初值}) \quad (\text{个外部脉冲})$$
- 该模式可省去软件中重装常数的语句，并可产生相当精确的定时时间，适合于作串行口波特率发生器。



三、模式 2 的应用举例

例6-2：当P3.4引脚（定时器T0的外部计数输入端）上的电平发生负跳变时，从P1.0输出一个500 μ s的同步脉冲。用查询方式编程实现该功能， $f_{osc}=6\text{MHz}$ 。

解：题目要实现的功能如图6-10所示。

（1）确定TMOD值

门控位GATE=0，高四位取0，选T0为模式2（M1M0=10），

首先为计数器工作方式（C/T=1）：（TMOD）=0000 0110B=06H

当P3.4引脚上的电平发生负跳变时，T0计数器加1，溢出标志TF0置1；

然后改变T0为500 μ s定时工作方式（C/T=0）：（TMOD）=02H

当上面TF0=1时使P1.0输出由1变为0。开始T0定时，到500 μ s产生溢出，使P1.0输出恢复高电平，T0又恢复外部事件计数方式。



(2) 计算T0初值

- T0工作在外部事件计数方式:

题目要求T0引脚出现一次外部事件时, 引起T0中断。

所以: 设计计数器初值为FFH, 当计数器再加1就溢出了。

即: $T0\text{初值} + 1 = 2^8$

$T0\text{初值} = 2^8 - 1 = 11111111B = 0FFH$

即: $(TH0) = (TL0) = 0FFH$

- T0工作在定时工作方式:

因为晶振频率为6MHz, 定时500 μ s。

所以: $(2^8 - T0\text{初值}) \times 2\mu s = 500\mu s$

得出: $T0\text{初值} = 6 = 06H,$

即: $(TH0) = (TL0) = 06H$

(3) 程序清单:

```
ORG    0000H
LJMP   START
ORG    0080H

START:  MOV    TMOD,#06H           ; 设置T0为模式2,外部计数方式
        MOV    TL0,#0FFH         ; T0计数器初值
        MOV    TH0,#0FFH
        SETB   TR0               ; 启动T0计数

LOOP1:  JBC     TF0, PTF01         ; 查询T0溢出标志, TF0=1时转, 且清TF0=0
        SJMP   LOOP1             ; 等待T0溢出

PTF01:  CLR     TR0               ; 停止计数
        MOV    TMOD,#02H         ; 设置T0为模式2, 定时方式
        MOV    TL0,#06H          ; 送初值, 定时500μs
        MOV    TH0,#06H
        CLR    P1.0              ; P1.0清0
        SETB   TR0              ; 启动定时500μs

LOOP2:  JBC     TF0,PTF02         ; 查询T0溢出标志, TF0=1时转并清TF0=0
        SJMP   LOOP2             ; 等待T0溢出中断 (定时时间到)

PTF02:  SETB    P1.0             ; P1.0置1
        CLR    TR0              ; 停止定时
        SJMP   START            ; 转向开始, 重新等待T0引脚的脉冲
```



例6-3：利用定时器T1的模式2对外部信号计数。要求每计满100次，将 P1.0端取反。

解：

(1) 选择模式

外部信号由T1（P3.5）引脚输入，每发生一次负跳变计数器加1，每输入100个脉冲，计数器发生溢出中断，中断服务程序将P1.0取反一次。

T1计数工作方式模式2的模式字为

$(\text{TMOD}) = 60\text{H}$ 。

T0不用时，TMOD的低4位可任取，但不能进入模式3，一般取0。



(2) 计算T1的计数初值

$$X = 2^8 - 100 = 156 = 9CH$$

因此:TL1的初值为9CH, 重装初值寄存器TH1=9CH。

(3) 程序清单

```
ORG    0000H
LJMP   MAIN

ORG    001BH      ; 中断服务程序入口
CPL    P1.0
RETI

MAIN:   MOV    TMOD,#60H ; 设置T1为模式2,外部计数方式
        MOV    TL1,#9CH  ; T1计数器初值
        MOV    TH1,#9CH
        MOV    IE, #88H  ; 定时器开中断
        SETB   TR1       ; 启动T1计数

HERE:   SJMP   HERE      ; 等待中断
```



§ 6.3.3 模式 3 及其应用

一、模式 3 的逻辑电路结构

工作模式3对T0和T1大不相同。

1、T0模式3的逻辑电路结构

2、T0模式3下T1的逻辑电路结构

二、模式 3 的应用举例



1、T0模式3的逻辑电路结构

- T0模式3的逻辑电路结构如图6-11所示。
- T0设置为模式3，TL0和TH0被分成两个相互独立的8位计数器。

- TL0:

用原T0的各控制位、引脚和中断源，即C/T，GATE，TR0，TF0，T0（P3.4）引脚，INT0（P3.2）引脚。

TL0可工作在定时器方式和计数器方式。其功能和操作与模式0、模式1相同（只是8位）。

- TH0:

只可用作简单的内部定时功能。

TH0占用了定时器T1的控制位TR1和T1的中断标志TF1，其启动和关闭仅受TR1的控制。



一、模式 3 的逻辑电路结构




图6-11 T0模式3结构-分成2个8计数器



2、T0模式3下T1的逻辑电路结构

- 定时器T1无工作模式3状态。在T0用作模式3时，T1仍可设置为模式0~2。
- T0模式3下T1的逻辑电路结构如图6-12所示。
- 由于TR1和TF1被定时器T0占用，计数器开关K已被接通，此时，仅用T1控制位，C/T切换其定时或计数器工作方式就可使T1运行。当计数器溢出时，只能将输出送入串行口或用于不需要中断的场合。一般作波特率发生器。



二、模式3的应用举例

例6-4：设某用户系统已使用了两个外部中断源，并置定时器T1工作在模式2，作串行口波特率发生器用。现要求再增加一个外部中断源，并由P1.0引脚输出一个5kHz的方波。Fosc=12MHz.

解：（1）工作模式

可设置T0工作在模式3计数器方式，把T0的引脚作附加的外部中断输入端，TL0的计数初值为FFH，当检测到T0引脚电平出现由1至0的负跳变时，TL0产生溢出，申请中断。T0模式3下，TL0作计数用，而TH0用作8位的定时器，定时控制P1.0引脚输出5kHz的方波信号。

（2）计算初值

- TL0的计数初值为FFH。

- TH0的计数初值X为： P1.0方波周期 $T=1/(5\text{kHz})=0.2\text{ms}=200\mu\text{s}$

用TH0作定时 $100\mu\text{s}$ 时， $X=256-100 \times 12/12=156$



(3) 程序清单

```
MOV  TMOD,#27H  ; T0为模式3,计数方式, T1为模式2, 定时方式
MOV  TL0,#0FFH  ; TL0计数初值
MOV  TH0,#156   ; TH0计数初值
MOV  TL1,#data   ; data是根据波特率, 要求设置的常数(初值)
MOV  TH1,#data
MOV  TCON,#55H  ; 外中断0, 外中断1边沿触发, 启动T0, T1
MOV  IE, #9FH   ; 开放全部中断
```

TL0溢出中断服务程序(由000BH转来)

```
TL0INT: MOV  TL0, #0FFH  ; TL0重赋初值
        (中断处理)
```

```
RETI
```

TH0溢出中断服务程序(由001BH转来)

```
TH0INT: MOV  TH0, #156   ; TH0重赋初值
        CPL   P1.0       ; P1.0取反输出
        RETI
```

串行口及外部中断0, 外部中断1的服务程序略



§ 6.3.4 综合应用举例

例6-5：设时钟频率为6MHz。编写利用T0产生1s定时的程序。

解：（1）定时器T0工作模式的确定

∵ 模式0最长可定时16.384ms； 模式1最长可定时131.072ms；

模式2最长可定时512 μs；

∴ 定时1s，可选用模式1，每隔100ms中断一次，中断10次从而达到1s的定时。

（2）求计数器初值X

$$\because (2^{16} - X) \times 12 / (6 \times 10^6) = 100 \times 10^{-3} \text{ s}$$

$$\therefore X = 15536 = 3CB0H$$

因此：(TL0)=0B0H (TH0)=3CH

（3）实现方法：对于中断10次计数，可使T0工作在计数方式，也可用循环程序的方法实现。本例采用循环程序法。




(4) 源程序清单:

	ORG	0000H	
	AJMP	MAIN	; 上电, 转向主程序
	ORG	000BH	; T0的中断服务程序入口地址
	AJMP	SERVE	; 转向中断服务程序
	ORG	0080H	; 主程序
MAIN :	MOV	SP,#60H	; 设堆栈指针
	MOV	B, #0AH	; 设循环次数
	MOV	TMOD,#01H	; 设置T0工作于模式1
	MOV	TL0,#0B0H	; 装计数值低8位
	MOV	TH0,#3CH	; 装计数值高8位
	SETB	TR0	; 启动定时
	SETB	ET0	; T0开中断
	SETB	EA	; CPU开中断
	SJMP	\$; 等待中断
SERVE:	MOV	TL0,#0B0H	; 重新赋初值
	MOV	TH0,#3CH	
	DJNZ	B,LOOP	; B-1不为0, 继续定时
	CLR	TR0	; 1s定时到, 停止T0工作 《例6-11完》
LOOP:	RETI		; 中断返回
	END		



门控位的应用

【例6-6】 测量 $\overline{\text{INT0}}$ 引脚上出现的正脉冲宽度，并将结果（以机器周期的形式）存放在70H和71H两个单元中。



- 将T0设置为方式1的定时方式，且GATE=1，计数器初值为0，将TR0置1。

- $\overline{\text{INT0}}$ 引脚上出现高电平时，加1计数器开始对机器周期计数。

- $\overline{\text{INT0}}$ 引脚上信号变为低电平时，停止计数。

- 然后读出TH0、TL0的值。



```
ORG    0000H
AJMP   MAIN
ORG    0200H
MAIN:  MOV    TMOD, #09H ;置T0为定时器方式1, GATE=1
        MOV    TH0, #00H ;置计数初值
        MOV    TL0, #00H
        MOV    R0, #70H ;置地址指针初值 (指向低字节)
L1:    JB      P3. 2, L1 ;高电平等待
        SETB   TR0      ;当INT0由高变低时使TR0=1, 准备好
L2:    JNB     P3. 2, L2 ;等待INT0变高
L3:    JB      P3. 2, L3 ;已变高, 启动定时, 直到INT0变低
        CLR    TR0      ;INT0由高变低, 停止定时
        MOV    @R0, TL0 ;存结果
        INC    R0
        MOV    @R0, TH0
        SJMP   $
END
```

【例】：设计实时时钟程序。时钟就是以秒、分、时为单位进行计时。用定时器与中断的联合应用。



解：（1）实现时钟计时的基本方法

① 计算计数初值。

时钟计时的最小单位是秒，可把定时器的定时时间定为100ms，计数溢出10次即得到1秒；10次计数可用软件方法实现。

假定使用定时器T0，以工作模式1进行100ms的定时。如 $f_{osc}=6\text{MHz}$ ，则计数初值X为：

$$\therefore X=15536=3\text{CB0H}$$

因此： (TL0)=0B0H (TH0)=3CH

② 采用定时方式进行溢出次数的累计，计满10次即得到秒计时。

设置软件计数器初值为10，每100ms定时时间到溢出中断，使软件计数器减1，直到减到0，则1s到。




③ 从秒到分和从分到时的计时是通过累计和数值比较实现的。

设置几个累加单元分别进行对1s, 1min, 1h进行计数。
满1s, 秒位累加; 满60s, 分位累加; 满60min, 时位累加; 满24h全部累加单元清0


(2) 程序流程及程序清单

① 主程序流程如图所示。





② 中断服务流程如图所示。



中断服务程序流程图



③ 源程序如下：

```
ORG 0000H
AJMP MAIN      ; 上电，转向主程序
ORG 001BH      ; T1的中断服务程序入口地址
AJMP SERVE     ; 转向中断服务程序
ORG 2000H      ; 主程序
MAIN: MOV SP,#60H ; 设堆栈指针
MOV TMOD,#10H  ; 设置T1工作于模式1
MOV 20H, #0AH  ; 设循环次数
CLR A
MOV 30H, A     ; 时单元清0
MOV 31H, A     ; 分单元清0
MOV 32H, A     ; 秒单元清0
SETB ET1       ; T1开中断
SETB EA        ; CPU开中断
MOV TL1,#0B0H  ; 装计数值低8位
MOV TH1,#3CH   ; 装计数值高8位
SETB TR1       ; 启动定时
SJMP $         ; 等待中断（可反复调用显示子程序）
```



中断服务程序:

SERVE: PUSH PSW	;保护现场
PUSH ACC	
MOV TL1,#0B0H	; 重新赋初值
MOV TH1,#3CH	
DJNZ 20H,RETUNT	; 1s未到, 返回
MOV 20H, #0AH	; 重置中断次数
MOV A, #01H	
ADD A, 32H	; “秒位” 加1
DA A	; 转换为BCD码
MOV 32H, A	
CJNE A, #60H, RETUNT	; 未满60s, 返回
MOV 32H, #00H	; 计满60s, “秒位” 清0
MOV A, #01H	
ADD A, 31H	; “分位” 加1
DA A	; 转换为BCD码
MOV 31H, A	
CJNE A, #60H, RETUNT	; 未满60min, 返回
MOV 31H, #00H	; 计满60min, “分位” 清0



```
MOV A, #01H
ADD A, 30H          ; “时位” 加1
DA A               ; 转换为BCD码
MOV 30H, A
CJNE A, #24H, RETUNT ; 未满足24h, 返回
MOV 30H, #00H      ; 计满24h, “时位” 清0
RETUNT: POP ACC     ; 恢复现场
POP PSW
RETI               ; 中断返回
END
```



§ 6.4 思考题与习题

- 1、定时器模式2有什么特点？适用于什么应用场合？
- 2、单片机用内部定时方法产生频率为100KHz等宽矩形波，假定单片机的晶振频率为12MHz。请编程实现。
- 3、89C51定时器有哪几种工作模式？有何区别？
- 4、89C51单片机内部设有几个定时器/计数器？它们是由哪些特殊功能寄存器组成？
- 5、定时器用作定时器时，其定时时间与哪些因素有关？作计数器时，对外界计数频率有何限制？



§ 6.4 思考题与习题

- 6、简述定时器四种工作模式的特点，如何选择和设定？
- 7、当定时器T0用作模式3时，由于TR1位已被T0占用，如何控制定时器T1的开启和关闭？
- 8、以定时器/计数器 1 进行外部事件计数。每计数1000个脉冲后，定时器T1转为定时工作方式。定时10ms后，又转为计数方式，如此循环不止。假定单片机晶振频率为6MHz，请使用模式1编程实现。
- 9、一个定时器的定时时间有限，如何实现两个定时器的串行定时，以满足较长定时时间的要求？
- 10、使用一个定时器，如何通过软、硬件结合的方法，实现较长时间的定时？



§ 6.4 思考题与习题

- 11、89C51定时器作定时和计数时，其计数脉冲分别由谁提供？
- 12、89C51定时器的门控制信号GATE设置为1时，定时器如何启动？
- 13、已知89C51单片机的 $f_{osc}=6\text{MHz}$ ，请利用T0和P1.0输出矩形波。矩形波高电平宽 $50\text{ }\mu\text{s}$ ，低电平宽 $300\text{ }\mu\text{s}$ 。
- 14、已知89C51单片机的 $f_{osc}=12\text{MHz}$ ，用T1定时。试编程由P1.0和P1.1引脚分别输出周期为 2ms 和 $500\text{ }\mu\text{s}$ 的方波。
- 15、单片机8031的时钟频率为 6MHz ，若要求定时值分别为 0.1ms ， 1ms ， 10ms ，定时器0工作在模式0、模式1和模式2时，其定时器初值各应是多少？



§ 6.4 思考题与习题

- 16、89C51单片机的定时器在何种设置下可提供三个8位计数器定时器？这时，定时器1可作为串行口波特率发生器。若波特率按9600b/s, 4800b/s, 2400b/s, 1200b/s, 600b/s, 100b/s来考虑，则此时可选用的波特率是多少（允许存在一定误差）？设 $f_{osc}=12\text{MHz}$ 。
- 17、试编制一段程序，功能为：当P1.2引脚的电平上跳时，对P1.1的输入脉冲进行计数；当P1.2引脚的电平下跳时，停止计数，并将计数值写入R6, R7。
- 18、设 $f_{osc}=6\text{MHz}$ 。试编制一段程序，功能为：对定时器T0初始化，使之工作在模式2，产生200 μs 定时，并用查询T0溢出标志的方法，控制 P1.0输出周期为2ms的方波。《完》



第六章结束